code:
;-------------------------------------------------
;クエスト定義ファイル
;QUEST_42:サンプルクエスト
;-------------------------------------------------
@QST_TITLE_QUEST_42
RESULTS '= TR("海岸の探索")

;ギルドメニューでクエストをフィルターする際のタグリスト。末尾の','を忘れないように
@QST_QUEST_TAG_42
RESULTS '= "周期,探索,"

;クエストメニュー右上段の概要欄。10行まで
@QST_SUMALLY1_QUEST_42
RESULTS:0 '= "領地から少し足を延ばしたところに、長大な海岸線が広がっている"
RESULTS:1 '= @"海の向こうから吹き付ける爽やかな海風が%ANAME(MASTER)%をくすぐる"
RESULTS:2 '= "風光明媚な景色をさかなに羽を伸ばしたい……が、ここも辺境の未開地であることに違いない"
RESULTS:3 '= "気を抜かずに探索するのが良いだろう"
RESULTS:5 '= "時には上陸してきた強大な海魔と出くわすこともあるかもしれない"
RESULTS:6 '= "敵わないと判断したならば退却するのが賢明だろう"

;クエストメニュー右下段の備考欄。難易度や条件、報酬等を記述。7行まで
@QST_SUMALLY2_QUEST_42
RESULTS:0 '= @"難易度：@C:{カラー_注意}@★×６～★×８@/C@"
RESULTS:1 '= "報酬：-　 備考:定期性クエスト(4期毎)"
RESULTS:2 '= "推奨パーティ人数:5人以上　　　推奨平均戦闘値：80以上"
RESULTS:3 '= @"@C:{カラー_選択不可}@「青い海！　照り付ける太陽！　旨い魚に水着の姉ちゃん！」@/C@"
RESULTS:4 '= @"@C:{カラー_選択不可}@「……こほん。……まぁなんだ、観光気分はほどほどにしておけよ」@/C@"
RESULTS:5 '= @"@C:{カラー_選択不可}@「ここは入り組んだ海岸で、入り江の奥は強力な海魔の巣だったりする」@/C@"
RESULTS:6 '= @"@C:{カラー_選択不可}@「観光気分で出て行った連中が帰ってこないなんてのはざらにある話だ」@/C@"

;-------------------------------------------------
;クエストの実行可能条件を定義
;-------------------------------------------------
@QST_AVAILABLE_QUEST_42(QUEST_ID)
#DIM QUEST_ID
;前提としてQUEST_6_湯泡郷の攻略が必要
SIF QST_SV_QUEST_FLAG:6 != 2
    RETURN 0
;クエストのクールタイム。完了時に加算され、期終了時に減算される
SIF QST_COOLTIME:QUEST_ID
    RETURN 0
RETURN 1

;-------------------------------------------------
;クエストのタイプ
;マップ探索を伴う場合 "MAP"
;グラフィカルなマップ探索を伴う場合 "GMAP"
;文章のみで完結する場合 "DIALOGUE"
;-------------------------------------------------
@QST_TYPE_QUEST_42
RESULTS '= "GMAP"

;-------------------------------------------------
;マップデータ定義
;[ROOM_ID:]　…　配置するエリアの種類
;[ROOM_PATH:]　…　エリア間の接続情報(※)
;[EVENT_LIST:]　…　エリアに配置するイベントのID(※)
;(※)要素の末尾の','を忘れないように注意
;EmueraをDEBUGMODEで起動した場合、[領地の手入れ]画面から編集したエリア情報をテキストファイルにエクスポートできる
;出力txtをベースに編集すると少し楽かもしれない
;-------------------------------------------------
@QST_SET_MAPDATA_QUEST_42(MAP_ID)
#DIM MAP_ID
;mapdata定義ファイルの格納先フォルダを指定
QST_GMAPDATA_PATH '= "ERB/QUEST/クエストデータ/QUEST_42_海岸の探索"
;マップデータ定義XMLとTXTのファイル名を指定。
;2ファイルの名称は拡張子を除いて一致している必要がある
QST_GMAPDATA_FILENAME '= "海岸の探索"
;このフラグを1にすると経路描画が行われなくなる。屋内マップなどあえて経路を図示する必要がない場合に立てるとよい
QST_HIDE_PATH = 0

;各エリア（ノード）に配置するイベントIDを記入。0-24はノード番号に対応
;EVENT_LIST末尾の’,’を忘れないように
;※ イベント戦闘があるので5,6(岬～灯台)と11,12(洞窟～巣窟)は探索状態を保持しない
_DIC_HO_MAPDATA:MAP_ID:0   '= "[EVENT_LIST:0,1,2,3,4,9,]" ;拠点
_DIC_HO_MAPDATA:MAP_ID:1   '= "[EVENT_LIST:0,10,18,19,]" ;ロングビーチ
_DIC_HO_MAPDATA:MAP_ID:2   '= "[EVENT_LIST:0,20,21,]" ;干潟の渡し
_DIC_HO_MAPDATA:MAP_ID:3   '= "[EVENT_LIST:0,30,35,]" ;海神の上顎
_DIC_HO_MAPDATA:MAP_ID:4   '= "[EVENT_LIST:0,40,46,47,48,]" ;難破船
_DIC_HO_MAPDATA:MAP_ID:5   '= "[EVENT_LIST:0,50,][NOT_SAVE_EXPLORED:1]" ;岬への道
_DIC_HO_MAPDATA:MAP_ID:6   '= "[EVENT_LIST:0,60,61,62,68,][NOT_SAVE_EXPLORED:1]" ;灯台跡
_DIC_HO_MAPDATA:MAP_ID:7   '= "[EVENT_LIST:0,]" ;悪魔の岩礁
_DIC_HO_MAPDATA:MAP_ID:8   '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:9   '= "[EVENT_LIST:0,91,]" ;マーメイドの入り江
_DIC_HO_MAPDATA:MAP_ID:10  '= "[EVENT_LIST:0,100,101,102,]" ;入り江
_DIC_HO_MAPDATA:MAP_ID:11  '= "[EVENT_LIST:0,110,][NOT_SAVE_EXPLORED:1]" ;入り江の洞窟
_DIC_HO_MAPDATA:MAP_ID:12  '= "[EVENT_LIST:0,120,121,][NOT_SAVE_EXPLORED:1]" ;海魔の巣窟
_DIC_HO_MAPDATA:MAP_ID:13  '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:14  '= "[EVENT_LIST:0,]"
_DIC_HO_MAPDATA:MAP_ID:15  '= "[EVENT_LIST:0,]"

;-------------------------------------------------
;クエスト固有の初期化処理
;-------------------------------------------------
@QST_INIT_QUEST_42
;クエスト固有のユニークエネミーをエネミー定義データに追加
CALL QST_QUEST_42_ENEMY_DATA
;クエストクールタイムを加算
QST_COOLTIME:QST_有効クエストID += 4

;-------------------------------------------------
;クエストの探索状態保持設定
;-------------------------------------------------
@QST_QUEST_42_SAVE_EXPLORED
RETURN 1

;-------------------------------------------------
;QST_INTRO_QUEST_n はQUEST開始前に呼ばれる
;クエストの導入文などを記述
;"DIALOGUE"型のクエストは当該関数のCALLのみ行う
;-------------------------------------------------
@QST_INTRO_QUEST_42
#DIM EVENT_FLAG_SKYLLA_NEST
#DIM EVENT_FLAG_SKYLLA_QUEEN

PRINTFORML %QST_LEADER_NAME()%は丘から眼下を望む
PRINTFORML 抜けるような青空の下、白波が砂浜に打ち付け、潮の香りが鼻をくすぐる
PRINTFORMW 長大な海岸線がうねるように遥か彼方まで続いている
PRINTFORMW 海岸線は時に内陸に鋭く切り込むように入り込んでおり、随所に深々と入り江をつくっている
PRINTL
PRINTFORML ギルドの情報によるとあの入り江の多くは強力無比な海魔の巣になっているようだ
PRINTFORMW 入り江に足を延ばすならば、十二分に警戒すべきだろう……

;このフラグが立っている場合ROOM_ID200(開始地点)で撤退コマンドが実行可能になる
FLAG:クエスト_中断を許可 = 1
;ゲーム内時刻を8:00にセット
HO_TIME = HO_CLOCK(800)
;0時にクエスト強制終了
FLAG:QST_終了時刻 = HO_CLOCK(2400)
CALL FL_PLAYBGM("Nostalgic_2.ogg")

;クエスト状況に合わせてルーム設定「探索状態を保持しない」を解除
;・「探索状態の保持」機能が追加される前にクエストクリアした人向けの処理
CALL QST_GET_SV_EVENT_FLAG(110)
EVENT_FLAG_SKYLLA_NEST = RESULT
CALL QST_GET_SV_EVENT_FLAG(120)
EVENT_FLAG_SKYLLA_QUEEN = RESULT
;「12.海魔の巣窟」に戦闘なしで移動できる状態なら
;「11.入り江の洞窟」と「12.海魔の巣窟」の設定を解除
;・入り江の洞窟で強行突破済 または スキュライベントクリア済(2.和解エンド, 9.根切りエンド)
IF EVENT_FLAG_SKYLLA_NEST || GROUPMATCH(EVENT_FLAG_SKYLLA_QUEEN, 2, 9)
    CALL QST_CLEAR_ROOM_NOT_SAVE_EXPLORED(11)
    CALL QST_CLEAR_ROOM_NOT_SAVE_EXPLORED(12)
ENDIF

;進捗フラグを1(既プレイ)に
QST_SV_QUEST_FLAG:QST_有効クエストID = 1

;-------------------------------------------------
;イベントリストに登録するイベント情報を記述する
;<記述方法>
;EVENT_TRIGGER
; "ENTER_ROOM" イベントが配置されたルームの入室時に強制的に実行される
; "EVENT_SLOT" イベントリストのクリック時に実行される
; "TRAIN_EXIT" 調教イベント終了後 ※予め変数:QST_TRAIN_EXIT_EVENT_ID に調教終了後に実行したいイベントIDを指定する必要がある
;EVENT_SLOT_NAME
; イベントリスト上の表示名
;EVENT_TYPE
; "ONCE" クエスト中一度のみ呼出されるイベント(※トリガーが"ENTER_ROOM"のイベントのみ機能)
;           実行済みかをフラグ管理するまでもないイベント(会話など)で使用する想定
;
;戻り値
; RESULT:0 = EVENT_TRIGGER
; RESULT:1 = EVENT_SLOT_NAME
; RESULT:2 = EVENT_TYPE
;-------------------------------------------------
;-------------------------------------------------
;EV0
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_42_EVENT_0
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
RESULTS:0 = %EVENT_TRIGGER%

;-------------------------------------------------
;実行するイベントの実行内容を記述
;イベント用の関数はTRYCALLFORMで呼ばれるため
;@QST_QUEST_{QUEST_ID}_EVENT_{EVENT_ID}の書式で記述する
;-------------------------------------------------
@QST_QUEST_42_EVENT_0
#DIMS ROOM_NAME

CALL HO_CREAR_LOG()
SELECTCASE HO_MASTER現在地
    CASE 0
        CALL HO_PRINT_LOG("街からは複雑に入り組んだ海岸線を一望することができる", 1)
    CASE 1
        CALL HO_PRINT_LOG("海岸に沿って白い砂浜が延々と続いている。この辺りは比較的安全なようだ", 1)
    CASE 2
        CALL HO_PRINT_LOG("水深の浅い干潟が対岸まで続いている。向こう岸まで歩いて渡れそうだ", 1)
    CASE 3
        CALL HO_PRINT_LOG("牙を思わせる鋭い岩礁が海面に林立している", 1)
    CASE 4
        CALL HO_PRINT_LOG("岩礁に乗り上げた帆船が打ち捨てられており、あちこちに積荷が散乱している", 1)
    CASE 5
        CALL HO_PRINT_LOG("突き出した岬の先端に向かって細くうねった道が続いている", 1)
    CASE 6
        CALL HO_PRINT_LOG("切り立った断崖の上にはかつての灯台跡がある。放棄されてから長い年月が過ぎているようだ", 1)
    CASE 7
        CALL HO_PRINT_LOG("奇怪な形の岩礁が海面から突き出している。酷く居心地の悪い雰囲気が辺りを包んでいる…", 1)
    CASE 9
        CALL HO_PRINT_LOG("海岸線は時に内陸に鋭く切り込むように入り込んでおり、随所に深々と入り江をつくっている", 1)
    CASE 10
        CALL HO_PRINT_LOG("海岸線は時に内陸に鋭く切り込むように入り込んでおり、随所に深々と入り江をつくっている", 1)
    CASE 11
        CALL HO_PRINT_LOG("半分は水没した洞窟がぽっかりと口を開けている", 1)
    CASE 12
        CALL HO_PRINT_LOG("生臭い魚の臭いが漂っており、足元には何かの骨が大量に散乱している", 1)
    CASEELSE
        CALL HO_CREAR_LOG()
ENDSELECT

;-------------------------------------------------
;EV10
;ロングビーチ
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_42_EVENT_10
; #region
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF
; #endregion

;EV18 海岸
@QST_ADD_EVENT_QUEST_42_EVENT_18
; #region
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "海岸"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_42_EVENT_18
#DIM EVENT_FLAG
#DIM CHARA_ID

;擬似設備の使用
;MODULE_ID:25(水場)を擬似的に使用している状態に
{
    CALL QST_CEV_USE_DUMMY_MODULE(
        @"[MODULE_ID:25]"
        +@"[ALLOW_COM:1]"
        +@"[MODULE_NAME:海岸]"
        +@"[USE_MODULE_MESSAGE:%QST_LEADER_NAME()%は波打つ海岸に近づいた]"
        +@"[MODULE_NAME_AT_USING:海岸を離れる]"
        +@"[STOP_USE_MODULE_MESSAGE:%QST_LEADER_NAME()%は波打つ海岸から離れた]"
        +@"[REDRESS_AT_STOP:1]"
        )
}

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT
;海岸使用中の場合EVENT_FLAGは1が立つ
IF RESULT
    ;コマンド実行後にイベント関数を呼ぶようEVENTCOMEND_CALLEEに追加
    MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE", "QST_QUEST_42_EVENT_18_SWIM_EVENT"
ELSE
    ;祭壇の使用をやめた時にイベント関数の呼び出しを削除
    MAP_REMOVE @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE"
ENDIF

; #endregion

;EV19 岩陰
@QST_ADD_EVENT_QUEST_42_EVENT_19
; #region
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "岩場の陰"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_42_EVENT_19
#DIM EVENT_FLAG
#DIM CHARA_ID

;擬似設備の使用
;MODULE_ID:28(岩場)を擬似的に使用している状態に
{
    CALL QST_CEV_USE_DUMMY_MODULE(
        @"[MODULE_ID:28]"
        +@"[PAUSE_BGM:1]"
        +@"[ALLOW_COM:1]"
        +@"[ALLOW_UFUFUCOM:1]"
        +@"[MODULE_NAME:岩場の陰]"
        +@"[USE_MODULE_MESSAGE:%QST_LEADER_NAME()%は岩場の陰に入った……]"
        +@"[MODULE_NAME_AT_USING:岩陰を出る]"
        +@"[STOP_USE_MODULE_MESSAGE:%QST_LEADER_NAME()%は岩陰を出た]"
        +@"[REDRESS_AT_STOP:1]"
        )
}

; #endregion

;-------------------------------------------------
;EV20
;干潟の渡し
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_42_EVENT_20
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_42_EVENT_20

CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
SIF RESULT
    RETURN
PRINTFORMW 眼前には水深の浅い干潟が対岸に向かって続いている
PRINTFORMW 一歩踏み出すと砂や珊瑚の堆積物が小気味の良い音を立てた
PRINTFORMW どうやらここから対岸まで歩いて渡ることができそうだ
PRINTL
PRINTFORML ただし地元の漁師たちの話によると、ここから先の入り江は海魔の巣窟であり、
PRINTFORMW 極めて危険なようだ。先へ進むなら準備は怠らない方が良いだろう……
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

;EV21_埋もれた宝箱
;region
@QST_ADD_EVENT_QUEST_42_EVENT_21
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "埋もれた宝箱"
#DIMS EVENT_TYPE = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_42_EVENT_21
#DIM CHARA_ID
CALL PRINT_TITLE("埋もれた宝箱", "QUEST")
PRINTFORMW %QST_LEADER_NAME()%は砂や珊瑚の堆積物の中から、宝箱が顔を覗かせていることに気がついた
PRINTFORMW しかし、完全にひっくり返る形で埋まってしまっており、掘り出すにはいささか困難を伴いそうだ……
PRINTL
PRINTFORMW どうしたものだろう？　掘り出しても良いし、放って置いても良い
PRINTL
CALL QST_ROLL_MULTI("身体,観察,","せっせと掘り出す(身体),使えるものを探す(観察),", "125,125",9)
CHARA_ID = RESULT:1
IF RESULT == 1
    IF RESULT:2 == 0
        PRINTFORMW %ANAME(CHARA_ID)%は宝箱の両脇に手を携え、力いっぱい引き抜こうと試みる
        PRINTFORMW 地面が盛り上がり、宝箱が堆積物の拘束から抜け始めた。あと一息だ……
        PRINTL
        PRINTFORMW さらに力を込めると、すっぽりと勢いよく引き抜けた！
    ELSEIF RESULT:2 == 1
        PRINTFORMW %ANAME(CHARA_ID)%は辺りを見回し、充分に丈夫で長い流木を見出した
        PRINTFORMW それを深々と差し込むと、てこの原理で宝箱を引き抜こうと試みる
        PRINTL
        PRINTFORMW 流木に体重を乗せぐっと力を込めると、宝箱は目論見通り持ち上がった！
    ENDIF
    PRINTFORMW ……宝箱の中には金貨や宝飾品が収められていた！
    CALL QST_CHANGE_ITEM(@"名誉点,20,MONEY,10000,{ITEM_金インゴット},2,")
    CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ELSEIF !RESULT
    PRINTFORMW 宝箱は深々と砂にはまり込んでおり、ちょっとやそっとでは動きそうにない
    PRINTFORMW %ANAME(CHARA_ID)%はその場で肩をすくめた
ENDIF
; #endregion

;-------------------------------------------------
;EV30 海神の上顎
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_42_EVENT_30
; #region
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_42_EVENT_30
CALL PRINT_TITLE("海神の上顎")
WAIT
PRINTFORMW 干潟を横切り、対岸へと渡り切ると%QST_LEADER_NAME()%は周囲の空気が変わったことを肌で感じ取った
PRINTFORMW 剣呑な空気が漂っており、林の木陰や岩陰、岩礁の裏から嫌な気配を感じ取った
PRINTL
PRINTFORMW いかなる脅威が待ち受けているかはわからないが、此処から先は充分に注意すべきだろう
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

;岩礁の横穴
@QST_ADD_EVENT_QUEST_42_EVENT_35
; #region
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "岩礁の横穴"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_42_EVENT_35
#DIM DYNAMIC CHARA_ID
#DIM DYNAMIC EVENT_FLAG

CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

CALL PRINT_TITLE("岩礁の横穴")
WAIT
PRINTFORMW 波打つ岩礁に沿って歩みを進めると、足元に人一人入れそうな程度の横穴を見出した
PRINTFORMW 海水に浸かっており、奥を見通すことは出来ない
PRINTFORMW だが%ANAME(QST_RANDCHARA())%はなぜかその横穴の奥から妙な気配を感じ取った
PRINTL
CALL ASK_YN("横穴に入る", "やめておく")
IF RESULT
    PRINTFORMW 気にはなるが同時に嫌な予感もする。やめておこう……
    RETURN
ENDIF
PRINTFORMW 松明を掲げた%QST_LEADER_NAME(1)%を先頭に横穴を奥へと進む
PRINTFORMW 横穴の奥へ踏み入ると、足元からぶよりとした、嫌な感触が伝わってきた……
PRINTL
PRINTFORMW 足元へ松明をかざすと、足元に大小のナメクジがびっしりと這い回っている！
PRINTL
PRINTFORMW ……いや、極彩色の体表を見るに、どうやらナメクジではなく、ウミウシの類の魔物のようだ
PRINTL
PRINTFORMW 魔物たちは光源に驚いたのか、触覚を伸ばし威嚇する構えを取っている
PRINTFORMW 進むにせよ退くにせよ、脅威を打ち払わねばならない……

;戦闘イベント初期化
CALL CC_INIT()
;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
CALL CC_ADDPARTY()
;エネミーを防御側で追加
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ヒュージ・スラッグ", "ヒュージ・シースラッグ１")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ヒュージ・スラッグ", "ヒュージ・シースラッグ２")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ヒュージ・スラッグ", "ヒュージ・シースラッグ３")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ヒュージ・スラッグ", "ヒュージ・シースラッグ４")
;戦闘イベント呼出
CALL CLOSE_COMBAT()

IF RESULT == 1
    CALL CC_APPLY_CHARA_HP
    CALL CC_GET_LOOT_FROM("DEFENDER")
    PRINTFORMW 次々と飛び寄ってくるシースラッグを打ち払いながら一行は進路を開く
    PRINTL
    PRINTFORML 小洞窟を奥まで進むと、最深部には小さな祠が一つ鎮座していた
    PRINTFORMW 異教の神を祀った祠のようだ。軟体の海生物を思わせる、なんともグロテスクな外見をしており、言葉にし難い生理的な嫌悪感を催させた
    ;初回成功時
    IF EVENT_FLAG < 2
        PRINTFORMW 祠には小箱が一つ置かれており、中には珊瑚を研磨して作られた小綺麗な腕輪が収められていた……
        ;レリック:珊瑚の腕輪を入手
        {
            CALL QST_GET_WEAPON(
                1
                ,"蜃気楼の腕輪"
                ,WPN_材質_その他
                ,3
                ,"[WPN_DECOY_AMPLIFIER:1]"
                ,""
                ,WPN_装備種別_レリック
                ,1
                )
        }
        CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 2)
    ELSE
        ;二度目以降
        PRINTFORMW %QST_LEADER_NAME()%は以前すでにこの横穴で腕輪を入手している
        PRINTFORML しかし誰が供えていったのか、幾つか宝飾品の類が祠に置かれている
        PRINTL
        CALL ASK_YN("供物を戦利品として持ち去る", "放っておきその場を立ち去る")
        IF !RESULT
            PRINTFORMW 戦利品としてありがたく頂戴していくとしよう
            CALL QST_CHANGE_ITEM(@"{ITEM_金属細工},4,{ITEM_金鉱石},2,")
        ELSE
            PRINTFORMW 触らぬ神に祟りなし、ともいう。放っておくとしよう
            CALL QST_CHANGE_ITEM("名誉点,20,")
        ENDIF
    ENDIF
ELSE
    CALL CC_APPLY_CHARA_HP
    CALL FL_PLAYBGM("Ambient_dark_2.mp3")
    PRINTFORML %QST_LEADER_NAME()%のふくらはぎに鈍い痛みが走る……
    PRINTFORMW 見るとシースラッグの毒刺胞が深々と突き刺さっている
    PRINTFORMW ……みるみるうちにしびれと目眩が%QST_LEADER_NAME()%を襲う。立っていられなくなりその場に膝をつく
    PRINTL
    PRINTFORMW 朦朧とする意識の中で、シースラッグの群体が身体に覆いかぶさってくるのを感じた
    PRINTFORMW ぬるぬるとぬめる不快な感触から逃れようとするも、身体がしびれ思うように動かない……
    PRINTL
    PRINTFORMW 軟体生物どもは何かを求めるように、しきりに身体の上を這い回っている
    PRINTFORMW 多くの魔物の例に漏れず、この生き物も獲物の体液からマナを接種するつもりのようだ……
    CALL QST_QUEST_42_DEFEAT_TRAIN_1
    CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
ENDIF

;イベント終了後にこのイベントを非活性に
CALL QST_DEACTIVATE_THIS_EVENT()

;EV32 敗北調教後イベント
;@QST_QUEST_6_DEFEAT_TRAIN_1終了後に呼ばれる
@QST_QUEST_42_EVENT_36
PRINTL
PRINTFORMW ……………………
PRINTFORMW ………………
PRINTFORMW …………
PRINTL
PRINTFORMW あれからどれくらいの時間が経過したのだろうか……
PRINTFORMW 洞窟内には潮が満ち始め、十二分に腹を満たしたシースラッグは水底に戻っていったようだ
PRINTFORMW %QST_LEADER_NAME()%は痺れの残る身体を起こし、よろよろと洞窟をあとにした

;汎用敗北イベントへ
CALL QST_QUEST_42_DEFEAT_EVENT

; #endregion

;-------------------------------------------------
;EV50 灯台跡
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_42_EVENT_50
; #region
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%

@QST_QUEST_42_EVENT_50
PRINTFORMW %QST_LEADER_NAME()%は岬の先端に向かって歩みを進める
PRINTFORMW すると頭上からぎゃあぎゃあと耳障りな叫びが響く
PRINTL
PRINTFORMW ――襲撃だ！

;戦闘イベント初期化
CALL CC_INIT()
;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
CALL CC_ADDPARTY()
;エネミーを防御側で追加
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ロックバード", "ロックバード１")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ロックバード", "ロックバード２")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "ロックバード", "ロックバード３")
;戦闘イベント呼出
CALL CLOSE_COMBAT()

;勝利
IF RESULT == 1
    ;戦闘で増減したステータスをキャラに反映
    CALL CC_APPLY_CHARA_HP()
    ;戦利品獲得処理
    CALL CC_GET_LOOT_FROM("DEFENDER")
    PRINTL
    PRINTFORML 怪鳥の群れを打ち払うことが出来た
    PRINTFORMW 先へ進むとしよう
    ;敗北
ELSE
    CALL CC_APPLY_CHARA_HP()
    PRINTL
    PRINTFORML 手ひどい損害を被ってしまった……ここは一度退くべきであろう
    PRINTFORMW %QST_LEADER_NAME()%は命からがら逃げ出した……！
    ;ゲームオーバー共通イベント
    CALL QST_QUEST_42_DEFEAT_EVENT
ENDIF

;-------------------------------------------------
;EV60 灯台跡
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_42_EVENT_60
; #region
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_42_EVENT_60
CALL PRINT_TITLE("灯台跡")
WAIT
PRINTFORMW %QST_LEADER_NAME()%は断崖沿いの道を、岬の先端へと歩みを進める
PRINTFORMW 終端にはかつての灯台跡と思われる構造物が残されていた
PRINTFORMW 建築様式から古帝国時代に打ち捨てられたものらしい
PRINTL
PRINTFORMW 崩落の具合を見るに、それは自然に倒壊したというより、中腹から強引にへし折られているように見えた
PRINTFORMW 周囲に散らばる石材には無数の穿孔が残されており、なにか凄まじい力を打ち付けた跡のようだ
PRINTL
PRINTFORMW だが、目下のところ周囲に目に付く脅威はなく、岸壁に打ち付けて砕ける波の音と風切音が響くのみである
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

@QST_ADD_EVENT_QUEST_42_EVENT_61
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "祭祀場跡"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_42_EVENT_61
#DIM DYNAMIC CHARA_ID

CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT == 1
    PRINTFORMW 地神を祀った祭壇がある
    PRINTFORMW 秋口には盛大な祭事が行われていたようだ
    RETURN
ENDIF

PRINTFORMW 灯台の周辺を探索していると古い祭祀場の跡を見つけた
PRINTFORMW どうやら灯台が出来るよりさらに以前に建てられたものらしい
PRINTFORMW それ以上の情報を得るには一定の知識がいりそうだ……
PRINTL
CALL QST_ROLL_MULTI("知識,","この地域の伝承について(知識),", "100", 12)
CHARA_ID = RESULT:1
IF RESULT == 1
    PRINTFORMW %ANAME(CHARA_ID)%によるとこれはこの地域の地神を祀った祭祀場の名残だそうだ
    HTML_PRINT "地神は天候を司り、この地の人々は秋の収穫の時期になると祭壇に歌や踊りを捧げ、次の一年の豊作と天候の安定を願った"
    PRINTFORMW 地神は見返りとしてこの地域に御使いを遣わしたという
    WAIT
    PRINTFORMW この御使いと住人の関係は良好で、恵みの雨をもたらす存在として大層重宝されていたようだ
    PRINTFORML
    PRINTFORMW ところが大戦を境にこの関係性は崩壊する
    PRINTFORML
    PRINTFORML 御使いは人であれ、魔物であれ、自らの領域に近づくものを見境なく襲うようになり
    PRINTFORMW 古来豊穣の象徴であった存在は破壊と災厄の象徴として恐れられるようになる
    PRINTFORMW そうしてついにはこの祭祀場に近寄るものは誰もいなくなったそうだ
    PRINTFORML
    PRINTFORMW その変心の理由は定かではない
    PRINTFORMW だが今日においても御使いは雲海を揺蕩っており、秋口になると嵐を伴って降りてくるのだと、信々深い地域の住人は口にしている
    PRINTFORML
    IF CHARA_ID == NAME_TO_CHARA("エル")
        CALL KJ_PRINT(CHARA_ID, "「このような話は珍しくないと聞く」", 1)
        CALL KJ_PRINT(CHARA_ID, "「私達のような研究者は大戦期において魔物を侵攻へと駆り立てた要因と同じなのではないかと考えているけれど、", 0)
        CALL KJ_PRINT(CHARA_ID, "御使いと称されるような神聖な存在と魔物を同一視することに抵抗を示す人々はとても多い」", 1)
        CALL KJ_PRINT(CHARA_ID, "「うーん……、こうなってしまうときちんと調査するのは……ちょっと難しいんだよね」", 1)
    ENDIF
    CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
ELSEIF !RESULT
    PRINTFORMW 特に何もわからない……。%ANAME(CHARA_ID)%はその場で肩をすくめた
ENDIF

@QST_ADD_EVENT_QUEST_42_EVENT_62
; #region
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "祭壇"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_42_EVENT_62
#DIM EVENT_FLAG
#DIM CHARA_ID

;擬似設備の使用
;MODULE_ID:205(小舞台)を擬似的に使用している状態に
{
    CALL QST_CEV_USE_DUMMY_MODULE(@"
        [MODULE_ID:205]
        [PAUSE_BGM:0]
        [ALLOW_COM:1]
        [MODULE_NAME:祭壇]
        [USE_MODULE_MESSAGE:%QST_LEADER_NAME()%は祭壇に立った……]
        [MODULE_NAME_AT_USING:祭壇を降りる]
        [STOP_USE_MODULE_MESSAGE:%QST_LEADER_NAME()%は祭壇を降りた]
        [REDRESS_AT_STOP:1]
        ")
}

CALL QST_GET_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT
;祭祀場使用中の場合EVENT_FLAGは1が立つ
IF RESULT
    ;コマンド実行後にボスバトル開始判定関数を呼ぶようEVENTCOMEND_CALLEEに追加
    MAP_SET @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE", "QST_QUEST_42_EVENT_60_BOSS"
ELSE
    ;祭壇の使用をやめた時にボスバトル開始判定関数の呼び出しを削除
    MAP_REMOVE @"QUEST_{QST_有効クエストID}_MAP", "EVENTCOMEND_CALLEE"
ENDIF

; #endregion

@QST_QUEST_42_EVENT_60_BOSS
#DIM DYNAMIC CHARA_ID
#DIM DYNAMIC EVENT_FLAG

CALL QST_GET_SV_EVENT_FLAG(62)
EVENT_FLAG = RESULT

IF !GROUPMATCH(SELECTCOM, T_N("演奏する"), T_N("踊る"))
    RETURN
ENDIF

IF HO_SEASON() !=HO_季節_秋 || EVENT_FLAG > 1
    PRINTFORMW %ANAME(MPLY)%は祭壇の上で歌声を披露してみたものの、歌声は波と風の音にかき消されて消えた
    PRINTFORMW 特に何かが起きる気配はない
    RETURN
ENDIF
CALL FL_STOPBGM()
IF EVENT_FLAG == 0
    PRINTFORMW %ANAME(MPLY)%が祭壇の上で歌声を披露すると、急に周囲の気温が下がったように思われた
    PRINTFORMW 雷鳴が轟き、暗雲が垂れこめる
    PRINTFORMW 岸壁に打ち付ける波は次第に激しくうねり、荒れ、高々と舞い上がった水飛沫と突風が顔を打つ
    PRINTFORML
    PRINTFORMW ――何かがいる！　不意に%ANAME(QST_RANDCHARA())%が叫び灯台の先で渦巻く黒雲を指差す
    PRINTFORMW 再び稲光が走る。一瞬の間ではあったが雲の中にいる何かの影を雷光が映し出した
    PRINTFORML
    PRINTFORMW 鳥か龍か、はっきりとした正体はわからない
    PRINTFORMW だが途方もなく巨大な何かが、黒雲の中を漂っているのは間違いない
    PRINTFORMW ……そして、こちらが怪物を捕捉したのと同時に、雲の中の何かもまたこちらを捕捉したようだ
    PRINTFORML
    PRINTFORMW 黒雲の中で幾つかの光球が瞬いたかと思えば、何条にも及ぶエネルギーの奔流が%QST_LEADER_NAME()%めがけ投射される！
    PRINTFORMW 間一髪のところで直撃を免れたものの、雷流の槍は石畳を容易く穿ち、大音響とともに爆ぜた
    PRINTFORML
    PRINTFORMW ――間違いない、雲の中の魔物こそ、噂に聞く怪物であろう！
    PRINTFORML
    PRINTFORMW %QST_LEADER_NAME()%は武器を携え脅威に対峙する。怪物は雲の中で悠然と宙返りすると、再び無数の雷球を生成し始めた
    PRINTFORMW 猛烈な風が吹き荒び、大粒の雨が顔を叩く
    PRINTFORML
    PRINTFORML
    PRINTFORMW 嵐の中で死闘が幕を開けようとしていた……

    ;戦闘イベント初期化
    CALL CC_INIT()
    ;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
    CALL CC_ADDPARTY()
    ;エネミーを防御側で追加
    CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_42_『雷嵐を携えし者』", "『雷嵐を携えし者』")
    CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_42_閃電器官", "閃電器官１")
    CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_42_閃電器官", "閃電器官２")
    CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_42_閃電器官", "閃電器官３")
    ;戦闘イベント呼出
    CALL CLOSE_COMBAT("[BGM:Boss_2.ogg][BGM_NOT_AUTOSTOP:1]")

    IF RESULT == 1
        ;戦闘で増減したステータスをキャラに反映
        CALL CC_APPLY_CHARA_HP()
        ;戦利品獲得処理
        CALL CC_GET_LOOT_FROM("DEFENDER")

        ;戦闘終了後
        CHARA_ID = QST_RANDCHARA(1)
        PRINTFORMW 電撃の波状攻撃をかわしつつ反撃を繰り返す%QST_LEADER_NAME()%にしびれを切らしたのか、ついに怪物は雲の合間から姿を現す
        PRINTFORMW それはマンタを想わせる扁平の魚類のような外見をしていた
        PRINTFORMW 長い尾を悠然と靡かせ、電雷を纏いながら%QST_LEADER_NAME()%に迫る
        PRINTFORML
        PRINTFORMW 頭部の両側から突き出た頭鰭が電気を帯び、激しくバチバチと火花を散らし明滅する
        PRINTFORMW 至近距離から電撃を浴びせ一挙に屠る算段らしい
        PRINTFORML
        PRINTFORMW ――しかし手の届く距離まで降りてきた事がかえって徒となった！
        PRINTFORMW 倒壊した灯台をよじ登った%ANAME(CHARA_ID)%が意を決して魔物の背に飛び乗る。そして頭頂部めがけ痛烈な一撃を放った
        PRINTFORML
        PRINTFORMW 頭部に致命的なダメージを受けた怪物は平たい身体をでたらめにばたつかせ、迷走する
        PRINTFORMW そして制御を失ったまま灯台の根元に激突すると、完全に絶命した
        PRINTFORMW
        CALL FL_STOPBGM()
        PRINTFORMW ……どうやら決着はついたようだ
        IF QST_PARTYNUM > 1
            PRINTFORMW %QST_LEADER_NAME()%は灯台の瓦礫の上でへたり込んでいる%ANAME(CHARA_ID)%に手を差しのべ、健闘を労う
        ENDIF
        PRINTFORML
        PRINTFORMW 激闘が終わる頃には嵐も止み、雲の合間から差し込んだ眩い陽光が%QST_LEADER_NAME()%のずぶ濡れの顔を照らした……
        PRINTFORML
        CALL FL_PLAYSE("Fanfare_0.ogg")
        PRINTFORMW ――『雷嵐を携えし者』を完全に討滅した！
        PRINTFORML
        CALL QST_CHANGE_ITEM("名誉点,1500,")
        CALL FL_PLAYBGM("Nostalgic_2.ogg")
        CALL QST_SET_SV_EVENT_FLAG(62, 1)
        ;宝箱イベントをアクティブに
        CALL QST_ACTIVATE_EVENT(68)
    ELSE
        ;戦闘で増減したステータスをキャラに反映
        CALL CC_APPLY_CHARA_HP()
        CALL FL_STOPBGM()
        PRINTL
        PRINTFORMW ――刃が立たない！
        PRINTFORMW 遥か天の高みから稲妻の一撃を加えてくる怪物に対して%QST_LEADER_NAME()%は始終防戦に徹せざるを得ない
        PRINTL
        PRINTFORMW しかし既に消耗は限界に達している。次の一撃を躱し切るのはおそらく不可能であろう
        PRINTFORMW ……潮時だ。そう決断した%QST_LEADER_NAME()%は撤退を決意した。命あらば再戦も可能であろう
        PRINTL
        ;ゲームオーバー共通イベント
        CALL QST_QUEST_42_DEFEAT_EVENT
    ENDIF
ELSEIF EVENT_FLAG == 1
    PRINTFORMW %ANAME(MPLY)%は再び祭壇の上で歌声を披露する
    PRINTFORML
    PRINTFORML 哀歌であったか、強者を称える讃歌であったか、あるいは単なる気分転換であったか
    PRINTFORMW いずれにせよ歌声は風に乗り虚空を漂い、打ち付ける波の音に混ざって消えた
    PRINTFORML
    PRINTFORML
    PRINTFORMW すると程なくして、歌声に呼応するように遥か頭上で遠雷が響く
    PRINTFORMW 突風が吹き、晴れているにもかかわらずぱらぱらと雨が振り始める
    PRINTFORMW 次第に風は強さを増し、顔を打つ雨露も大きさを増していく
    PRINTFORML
    PRINTFORMW 不意に耳をつんざくような轟音とともに雷が落ち、石畳を激しく打った
    PRINTFORMW %ANAME(QST_RANDCHARA())%は反射的に身を屈める
    PRINTFORML
    PRINTFORMW 体勢を立て直すと、眼前には無数の黄金色に煌めく粒子が漂っている
    PRINTFORMW それらは宙を優雅に舞い、踊り、凝集し一つの形をなした
    PRINTFORML
    PRINTFORMW 扁平な魚類を思わせるそのシルエットは、かつてこの場で屠った嵐の主、その似姿のように思われた
    PRINTFORML
    PRINTFORML 以前と決定的に異なるのは、再び相まみえた嵐の主はただ優雅に宙を漂うだけであり
    PRINTFORMW かつてのような攻撃的な意志は全く感じられないことであった
    PRINTFORML
    PRINTFORMW 雲の合間から陽光が差し込み、陽に照らされた雨が輝きを纏って降り注ぐ
    PRINTFORMW どこか神秘的な光景の中で、嵐の主は挑発的にくるりと宙返りしてみせた
    PRINTFORML
    PRINTFORMW ……なぜそう感じたかはわからないが、それはまるで「かかってこい」と言わんばかりだ
    PRINTFORMW しかし、敵意こそ感じないものの、眼前の存在が放つ気配は途方もなく力に満ち満ちている
    PRINTFORML
    PRINTFORMW ……生半可な力では敵わないだろう
    PRINTFORML
    PRINTFORML
    PRINTFORMW ――この戦いは、おそらく勝利したところで名誉以外に得るものはなにもない
    PRINTFORMW あなたは眼の前の存在に対し武勇を示してもいい。無論益のない戦いを避けるのも懸命な判断だ
    PRINTFORML
    PRINTFORMW その判断は%QST_LEADER_NAME(1)%に委ねられている
    PRINTFORML
    CALL ASK_YN("高々と武器を掲げる", "その場を立ち去る")
    IF RESULT
        PRINTFORML
        PRINTFORMW あえて利の薄い戦いで消耗する必要は無いと判断した%QST_LEADER_NAME()%祭壇を降り、その場を立ち去った
        RETURN
    ENDIF
    PRINTFORML
    PRINTFORMW 意地か、あるいは功名心か、はたまた闘争心の発露か、%QST_LEADER_NAME(1)%は武器を天上の存在に向け突きつける
    PRINTFORML
    PRINTFORMW 嵐の主はまるで歓喜を示すかのように天に大きく輪を描くと、高度を下げ、一度吠えた
    PRINTFORML
    PRINTFORMW 咆声に呼応するように天雷が響き、激しく叩きつけるような風雨が%QST_LEADER_NAME()%を襲う
    PRINTFORML
    PRINTFORMW ……%QST_LEADER_NAME(1)%は武器を手にしたことを少し後悔したかもしれない
    PRINTFORMW なぜなら眼の前の存在が放つ威圧感は、以前敵対したときよりも比較にならないほどに重く、強い
    PRINTFORML
    PRINTFORML
    PRINTFORMW %ANAME(QST_RANDCHARA())%を指向した嵐の主の、巨大な両胸鰭がきらりと明滅する
    PRINTFORML
    PRINTFORMW ――次の瞬間、槍の如く巨大な棘が二本、知覚することがほぼ不可能な速度で放たれる！
    PRINTFORMW それは衝撃波を纏って標的をかすめると、灯台の基部に深々と突き刺さった
    PRINTFORMW 轟音とともに電雷が爆ぜ、僅かに形を残していた灯台は音を立てて完全に倒壊した
    PRINTFORML
    PRINTFORML
    PRINTFORMW ……相手は人智を超えた存在だ。敵意こそないとは言え油断すれば即座に命を落とすだろう
    PRINTFORMW だが%QST_LEADER_NAME()%とて易々とやられてやる道理はない。冒険者には冒険者の矜持というものがある
    PRINTFORML
    PRINTFORMW 吹き荒ぶ嵐の中で激闘が幕を開けた……
    ;戦闘イベント初期化
    CALL CC_INIT()
    ;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
    CALL CC_ADDPARTY()
    ;エネミーを防御側で追加
    CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_42_『ストームルーラー』", "『ストームルーラー』")
    CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_42_閃電器官", "閃電器官１")
    CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_42_閃電器官", "閃電器官２")
    CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_42_閃電器官", "閃電器官３")
    CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_42_電磁投射刺胞", "電磁投射刺胞・左翼")
    CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_42_電磁投射刺胞", "電磁投射刺胞・右翼")
    ;戦闘イベント呼出
    CALL CLOSE_COMBAT("[BGM:Boss_1.ogg][BGM_NOT_AUTOSTOP:1]")

    IF RESULT == 1
        ;戦闘で増減したステータスをキャラに反映
        CALL CC_APPLY_CHARA_HP()
        ;戦利品獲得処理
        CALL CC_GET_LOOT_FROM("DEFENDER")

        ;戦闘終了後
        CHARA_ID = QST_RANDCHARA(1)
        PRINTFORMW 雷が石畳を打ち、瞬時に蒸発した水分が靄のように立ち込める
        PRINTFORMW 閃光が迸り、砕け散った石片が宙を舞う
        PRINTFORMW 目にも止まらぬ速度で射出される剣状棘が、衝撃波を伴って襲い来る
        PRINTFORML
        PRINTFORMW 激烈な攻撃の嵐を縫い、%QST_LEADER_NAME()%は反撃の機会を伺う
        PRINTFORML
        PRINTFORMW 戦いは消耗戦の様相を呈した。嵐の主の大仰な攻撃に対し、%QST_LEADER_NAME()%の正確無比な反撃が一つ、また一つと攻撃手段を奪う
        PRINTFORMW 圧倒的な破壊力を持つ左右の刺胞が失われると、嵐の主は甲高い鳴き声を上げた
        PRINTFORMW その音色は何処か愉快そうな、笑い声のように聞こえた
        PRINTFORML
        PRINTFORMW 激しい攻撃の応酬の末、双方ともに切れる手札を全て切り尽くす
        PRINTFORMW 戦いの果に両者は祭壇上で相対すると、ほぼ真正面から激突した
        PRINTFORML
        PRINTFORML
        PRINTFORMW %QST_LEADER_NAME()%の渾身の一撃が嵐の主を討った
        PRINTFORML
        CALL FL_STOPBGM()
        PRINTFORMW ……どうやら決着はついたようだ
        PRINTFORML
        PRINTFORMW 嵐の主を体躯を構成する光の粒子が爆ぜ、あたりにきらきらと舞い落ちる
        PRINTFORML
        PRINTFORMW 精根尽き果てへたり込む%ANAME(QST_RANDCHARA())%の頭上で、長く尾を引くような満足げな咆哮が響いた……
        PRINTFORML
        CALL FL_PLAYSE("Fanfare_0.ogg")
        PRINTFORMW ――『ストームルーラー』を完全に討滅した！
        PRINTFORML
        CALL QST_CHANGE_ITEM("名誉点,2000,")
        CALL FL_PLAYBGM("Nostalgic_2.ogg")
        CALL QST_SET_SV_EVENT_FLAG(62, 2)
    ELSE
        ;戦闘で増減したステータスをキャラに反映
        CALL CC_APPLY_CHARA_HP()
        CALL FL_STOPBGM()
        PRINTL
        PRINTFORMW ――まるで刃が立たない！
        PRINTFORMW 以前の戦いを遥かに凌ぐ破壊力の攻撃は、%QST_LEADER_NAME()%に一切の付け入る隙を与えなかった
        PRINTL
        PRINTFORMW 退却を試みる%QST_LEADER_NAME()%に対し、嵐の主は追撃を加えようとしなかった
        PRINTFORMW その代わりに、宙を旋回すると何処か不服げに一度鳴き、雲上に姿を消した……
        PRINTL
        ;ゲームオーバー共通イベント
        CALL QST_QUEST_42_DEFEAT_EVENT
    ENDIF
ENDIF

@QST_ADD_EVENT_QUEST_42_EVENT_68
; #region
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "豪奢な宝箱"
;このイベントは初期状態で非アクティブ化
;『雷嵐を携えし者』撃破時にアクティブに
CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RETURN

@QST_QUEST_42_EVENT_68
#DIM EVENT_FLAG
#DIM CHARA_ID

PRINTFORMW %QST_LEADER_NAME()%は怪物が激突して倒壊した灯台の根本に綺羅びやかな宝箱が埋もれているのを見出した
PRINTFORMW 蓋を開けてみると、儀式のために用いられたと見られる宝飾が収められていた
PRINTL
PRINTFORMW その中でひときわ目を引いたのは細微な装飾の施された銀の弓であった
PRINTFORMW 手に取るとほのかに温かく、弦を引くと大変しなやかにしなった
PRINTL
PRINTFORMW 祭祀用に製作された弓のようだが、実用に充分耐える品のようだ
PRINTFORMW 試しに付近の樹に狙いを定め、矢を放つ
PRINTL
PRINTFORMW すると放たれた矢は激しく明滅し、火花を散らしながら空を切る
PRINTFORMW そして樹に突き刺さるや樹皮を鋭く、深く抉るように爆ぜた

;武器:召雷の祭祀弓を入手
{
    CALL QST_GET_WEAPON(
        15
        ,"『召雷の祭祀弓』"
        ,WPN_材質_銀
        ,3
        ,""
        ,"[WPN_ELECTRIC_ARROW:1]"
        ,WPN_装備種別_武器
        ,1
        )
}
CALL QST_DEACTIVATE_THIS_EVENT()

@QST_QUEST_42_EVENT_10
CALL PRINT_TITLE("どこまでも続く白浜")
WAIT
PRINTFORML %QST_LEADER_NAME()%の眼前には延々と続く白浜が広がっている
PRINTFORMW 燦々たる陽光の下、穏やかな波が寄せては引き、海鳥の鳴き声が辺りに響いている
PRINTFORMW 冒険に疲れたのならば浜辺で穏やかな時間を過ごすのも悪くない
PRINTL
PRINTFORMW あるいは……熾烈な冒険を求めるならば眼前の白浜を越え、干潟の先の対岸を目指すのも良いだろう
PRINTFORML 地元の漁師たちの談によると、対岸の入り江は海獣海魔の巣窟であり近づく者は殆どいないという
PRINTFORML
PRINTFORMW 未踏の地であるがゆえに、人しれぬ財宝が眠っているかもしれない
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
; #endregion

;-------------------------------------------------
;以下イベントリストで管理する必要のない処理,イベント固有の関数など
;-------------------------------------------------
;汎用敗北イベント
@QST_QUEST_42_DEFEAT_EVENT
PRINTL
PRINTL
PRINTFORMW %QST_LEADER_NAME()%はボロボロの身を引きずってなんとか街まで帰還した……
PRINTFORML
PRINTFORMW ……今回の探索は失敗に終わってしまった。帰還して一度態勢を立て直すことにしよう
PRINTL
CALL PRINT_NOTICE("治療費や宿泊費として15,000ディナリを失った……")
MONEY -= 15000
QST_SV_QUEST_FLAG:QST_有効クエストID = 1
TFLAG:クエスト_完了フラグ = 1
CALL TRAIN_EXIT()

;汎用敗北イベント
@QST_QUEST_42_DEFEAT_EVENT_2
PRINTL
PRINTFORMW %QST_LEADER_NAME()%は街の宿で目を覚ました。一体いつ宿に戻って来たのか、記憶が判然としない
PRINTFORML
PRINTFORMW 身体はぐっしょりと汗ばみ、節々に痛みを感じる
PRINTFORML
PRINTFORMW 「ようやく起きたのかい？　驚いたよ。ふらふらと帰ってきたかと思えば三日三晩寝込んじまうんだもの」
PRINTFORML
PRINTFORMW 一体いつ出発し、いつ戻ってきたのかが思い出せない。記憶が酷く混濁している
PRINTFORML
PRINTFORMW ……しかし下腹部に残る妙な感触だけが、確かに何かただならぬ事があったことを記憶している
PRINTL
CALL PRINT_NOTICE("宿泊費として15,000ディナリを失った……")
MONEY -= 15000
QST_SV_QUEST_FLAG:QST_有効クエストID = 1
TFLAG:クエスト_完了フラグ = 1
CALL TRAIN_EXIT()

;-------------------------------------------------
;シースラッグ敗北調教
;-------------------------------------------------
@QST_QUEST_42_DEFEAT_TRAIN_1
#DIM CHARA_ID
#DIM CONST 派生イベントID = 36

FLAG:ウフフフラグ = 1
FLAG:クエスト_刻限による終了無効 = 1
FLAG:調教モード = 調教_慰安
TFLAG:調教ターン数 = 18

;スキップ許可
FLAG:調教スキップ可 = 1

;主導除外フラグが立ったキャラクターは慰安調教で常にターゲットに回る。
;拘束中など自主的に行動させたくないシチュエーションでこのフラグを立てる。
CVARSET CFLAG, GETNUM(CFLAG, "調教参加フラグ") , 0
FOR LOCAL, 0, QST_PARTYNUM
    CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
    ;調教コマンドのプレイヤーにならなくなる。一方的に嬲られるシチュ用
    CFLAG:CHARA_ID:主導除外 = 1
    ;良感情を極めて得にくくなるフラグ
    CFLAG:CHARA_ID:凌辱被害者フラグ = 1
    CFLAG:CHARA_ID:調教参加フラグ = 1
NEXT

;慰安モブを追加
;・シースラッグ×5
CALL QST_QUEST_42_SET_TRAIN_MOB(0, 0, 0, 5)

CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(派生イベントID)

RETURN

; #endregion

;-------------------------------------------------
; 海岸で水浴びした場合に発生するサブイベント
;-------------------------------------------------
@QST_QUEST_42_EVENT_18_SWIM_EVENT
#DIM  DYNAMIC CHARA_ID
#DIM  DYNAMIC EVENT_FLAG
#DIMS DYNAMIC LIST_CHARA
#DIMS DYNAMIC 衣装名

IF !GROUPMATCH(SELECTCOM, T_N("水浴び"))
    RETURN
ENDIF

IF HO_SEASON() !=HO_季節_夏
    PRINTFORML %ANAME(MPLY)% 海水浴をするにはいささか肌寒い…
    RETURN
ENDIF

;水着を着用してるキャラを抽出してイベント対象キャラの候補リストをつくる
FOR LOCAL, 0, MPLY_NUM
    CHARA_ID = MPLY:LOCAL
    {
        IF C_COND("衣装属性:胸:水着",,CHARA_ID) && !C_COND("露出部位:胸",,CHARA_ID)
            || C_COND("衣装属性:秘部:水着",,CHARA_ID) && !C_COND("露出部位:秘部",,CHARA_ID)
    }
        LIST_CHARA = %LIST_ADD(LIST_CHARA, TOSTR(CHARA_ID))%
    ENDIF
NEXT
FOR LOCAL, 0, MTAR_NUM
    CHARA_ID = MTAR:LOCAL
    {
        IF C_COND("衣装属性:胸:水着",,CHARA_ID) && !C_COND("露出部位:胸",,CHARA_ID)
            || C_COND("衣装属性:秘部:水着",,CHARA_ID) && !C_COND("露出部位:秘部",,CHARA_ID)
    }
        LIST_CHARA = %LIST_ADD(LIST_CHARA, TOSTR(CHARA_ID))%
    ENDIF
NEXT

IF LIST_COUNT(LIST_CHARA) == 0
    PRINTFORML 夏の暖かい風が%ANAME(MPLY)%の肌を撫でる
    PRINTFORML 水着を着ていれば海水浴をするのも悪くはないかもしれない
    RETURN
ENDIF

CHARA_ID = TOINT(LIST_RAND(LIST_CHARA))
PRINTFORML %ANAME(CHARA_ID)%は海水に浸かり、ひと泳ぎすることにした
PRINTL
SELECTCASE RAND:10
    CASE 0 TO 2
        PRINTFORML ――とても気持ちがいい！
    CASE 3 TO 7
        IF C_COND("衣装属性:秘部:水着",,CHARA_ID)
            衣装名 = %C_COND_MAP_GETS(CHARA_ID, "被覆衣装描写_秘部")%
        ELSE
            衣装名 = %C_COND_MAP_GETS(CHARA_ID, "被覆衣装描写_胸")%
        ENDIF
        PRINTFORML ……おや？ 泳いでいるうちに%ANAME(CHARA_ID)%は%衣装名%に何かが引っかかっていることに気がついた
        SELECTCASE RAND:6
            CASE 0
                CALL QST_CHANGE_ITEM(@"{ITEM_ディナリア銀貨},1,")
            CASE 1
                CALL QST_CHANGE_ITEM(@"{ITEM_鉄インゴット},1,")
            CASE 2
                CALL QST_CHANGE_ITEM(@"{ITEM_ポーション},1,")
            CASE 3
                CALL QST_CHANGE_ITEM(@"{ITEM_銀鉱石},1,")
            CASE 4
                CALL QST_CHANGE_ITEM(@"{ITEM_バイブ},1,")
            CASE 5
                CALL QST_CHANGE_ITEM(@"RELIC_赤べこ,1,")
        ENDSELECT
    CASE 8 TO 9
        IF C_COND("衣装属性:秘部:水着",,CHARA_ID)
            衣装名 = %C_COND_MAP_GETS(CHARA_ID, "被覆衣装描写_秘部")%
            CALL HO_STRIP(CHARA_ID, 被覆部位_秘部, 衣装状態_脱, 1)
        ELSE
            衣装名 = %C_COND_MAP_GETS(CHARA_ID, "被覆衣装描写_胸")%
            CALL HO_STRIP(CHARA_ID, 被覆部位_乳首, 衣装状態_脱, 1)
        ENDIF
        PRINTFORMW 不意に押し寄せた高波が%ANAME(CHARA_ID)%の身体を打つ！
        PRINTFORMW なんとか体勢を立て直したものの、%ANAME(CHARA_ID)%は妙な開放感を感じる
        CALL PRINT_NOTICE(@"いつの間にか波にさらわれていた%ANAME(CHARA_ID)%の%衣装名%が水面を漂っている……", カラー_ピンク)
ENDSELECT

RETURN

;-------------------------------------------------
;@QST_QUEST_42_ENEMY_DATA
; クエスト固有エネミーの読み込み処理
; クエストで共通的に使い回すエネミーの戦闘データは@CC_ENEMY_DATAに記述されているが
; クエスト側で後付で追加することもできる。(@QST_INIT_QUEST_XXなどで明示的に読込み処理を呼び出すこと)
; 特定のクエストでしか使わないエネミーのデータ（ボスなど）はクエスト側に記述するとよいだろう
; パラメータの詳細は @CC_ENEMY_DATA 参照
; ！！！ENEMY_DEF_NAMEはユニークである必要があるためクエスト側でエネミーを追加する場合はENEMY_DEF_NAMEに'QUEST_XX_'を接頭すること！！！
;-------------------------------------------------
@QST_QUEST_42_ENEMY_DATA
{
    DT_ROW_ADD "ENEMY_DATA"
        ,"ENEMY_DEF_NAME", "QUEST_42_『雷嵐を携えし者』"
        ,"MAX_HP", 18000
        ,"STR", 180
        ,"MAG", 100
        ,"ATTACK", 180
        ,"DEFENCE", 45
        ,"ARMOR", 450
        ,"AGILITY", 80
        ,"HATE_RATE", 100
        ,"TARGETING", WPN_脅威選定_無作為
        ,"REGIST_POISON", 70
        ,"REGIST_STUN", 80
        ,"REGIST_CHARM", 70
        ,"REGIST_PANIC", 70
        ,"REGIST_BLIND", 70
        ,"REGIST_DEATH", 100
        ,"ENEMY_PASSIVE_SKILL", "[ATK_ADD_CAST_TOKEN:1]"
        ,"ENEMY_ACTIVE_SKILL", "[ACT_THUNDER_STORM:2]"
        ,"DROP_ITEM_LIST", @"MONEY,20000,1000,{ITEM_技巧の宝玉},2,1000,{ITEM_魔獣の心核},1,1000,"
        ,"EXP", 12
}
{
    DT_ROW_ADD "ENEMY_DATA"
        ,"ENEMY_DEF_NAME", "QUEST_42_『ストームルーラー』"
        ,"MAX_HP", 38000
        ,"STR", 280
        ,"MAG", 100
        ,"ATTACK", 180
        ,"DEFENCE", 45
        ,"ARMOR", 450
        ,"AGILITY", 200
        ,"HATE_RATE", 100
        ,"TARGETING", WPN_脅威選定_無作為
        ,"REGIST_POISON", 70
        ,"REGIST_STUN", 80
        ,"REGIST_CHARM", 70
        ,"REGIST_PANIC", 70
        ,"REGIST_BLIND", 70
        ,"REGIST_DEATH", 100
        ,"ENEMY_PASSIVE_SKILL", "[ATK_ADD_CAST_TOKEN:1]"
        ,"ENEMY_ACTIVE_SKILL", "[ACT_THUNDER_STORM:3]"
        ,"DROP_ITEM_LIST", @"MONEY,20000,1000,{ITEM_金インゴット},5,1000,{ITEM_技巧の宝玉},1,1000,{ITEM_魔獣の心核},1,1000,"
        ,"EXP", 16
}
{
    DT_ROW_ADD "ENEMY_DATA"
        ,"ENEMY_DEF_NAME", "QUEST_42_閃電器官"
        ,"MAX_HP", 4000
        ,"STR", 80
        ,"MAG", 125
        ,"ATTACK", 80
        ,"DEFENCE", 25
        ,"ARMOR", 300
        ,"AGILITY", 60
        ,"HATE_RATE", 100
        ,"TARGETING", WPN_脅威選定_無作為
        ,"REGIST_POISON", 30
        ,"REGIST_STUN", 30
        ,"REGIST_CHARM", 70
        ,"REGIST_PANIC", 70
        ,"REGIST_BLIND", 70
        ,"REGIST_DEATH", 100
        ,"ENEMY_PASSIVE_SKILL", "[ATK_ADD_CAST_TOKEN:1]"
        ,"ENEMY_ACTIVE_SKILL", "[ACT_CAST_SMITE:2][ACT_SHOCK_BOLT:2]"
        ,"DROP_ITEM_LIST", @""
        ,"EXP", 6
}
{
    DT_ROW_ADD "ENEMY_DATA"
        ,"ENEMY_DEF_NAME", "QUEST_42_電磁投射刺胞"
        ,"MAX_HP", 6500
        ,"STR", 450
        ,"MAG", 80
        ,"ATTACK", 450
        ,"DEFENCE", 40
        ,"ARMOR", 300
        ,"AGILITY", 60
        ,"HATE_RATE", 100
        ,"TARGETING", WPN_脅威選定_無作為
        ,"REGIST_POISON", 0
        ,"REGIST_STUN", 0
        ,"REGIST_CHARM", 70
        ,"REGIST_PANIC", 70
        ,"REGIST_BLIND", 70
        ,"REGIST_DEATH", 100
        ,"ENEMY_PASSIVE_SKILL", "[ATK_ADD_CAST_TOKEN:1]"
        ,"ENEMY_ACTIVE_SKILL", "[ACT_CAST_SMITE:3][ACT_SHOCK_BOLT:3]"
        ,"DROP_ITEM_LIST", @""
        ,"EXP", 6
}
{
    DT_ROW_ADD "ENEMY_DATA"
        ,"ENEMY_DEF_NAME", "QUEST_42_『デミスキュレー・クイーン』"
        ,"MAX_HP", 10000
        ,"STR", 75
        ,"MAG", 70
        ,"ATTACK", 75
        ,"DEFENCE", 40
        ,"ARMOR", 400
        ,"AGILITY", 80
        ,"HATE_RATE", 100
        ,"TARGETING", WPN_脅威選定_無作為
        ,"REGIST_POISON", 20
        ,"REGIST_STUN", 60
        ,"REGIST_CHARM", 70
        ,"REGIST_PANIC", 70
        ,"REGIST_BLIND", 70
        ,"REGIST_DEATH", 100
        ,"ENEMY_PASSIVE_SKILL", "[ATK_ADD_CAST_TOKEN:1]"
        ,"ENEMY_ACTIVE_SKILL", "[ACT_CAST_ATTACK_ALL:2]"
        ,"DROP_ITEM_LIST", @"MONEY,20000,1000,{ITEM_淫堕の宝玉},3,1000,{ITEM_高効率魔導触媒},2,1000,"
        ,"EXP", 12
}


code:
;-------------------------------------------------
;EV0 拠点
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_42_EVENT_1
; #region
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "街の宿"

RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_42_EVENT_1
#DIM EVENT_FLAG
#DIM TARGET_ID

PRINTL
PRINTFORMW 探索の拠点としてお誂え向きな小さな宿屋がある
PRINTFORML 機会は一度きりだが、身体を休めることもできそうだ
PRINTFORMW ただし代金として2,000ディナリ必要だ
CALL ASK_YN("休憩する", "やめておく")
IF !RESULT
    IF MONEY < 2000
        PRINTFORMW お金が足りない……
        PRINTFORMW %QST_LEADER_NAME()%は財布を開きため息をついた
        RETURN
    ENDIF
    PRINTFORMW %QST_LEADER_NAME()%は部屋を借り少し休憩することにした……
    ;パーティ全員の体力を回復
    FOR LOCAL, 0, QST_PARTYNUM
        TARGET_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
        CALL QST_CHANGE_HP(TARGET_ID, 2500)
    NEXT
    CALL QST_RESET_SKILL_COOLTIME(1)
    CALL QST_CHANGE_ITEM("MONEY, 2000")
    WAIT
    CALL QST_DEACTIVATE_THIS_EVENT()
ELSE
ENDIF
; #endregion

;-------------------------------------------------
;EV0 会話系イベント
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_42_EVENT_2
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "宿泊客"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_42_EVENT_2
PRINTFORMW 「あんたら冒険者だろう？　宝探しに夢中になるのはいいが、岬の方には近づかねぇこった」
PRINTFORMW 「あそこは昔からとんでもなくおっかねぇ魔物が住み着いてて、まぁよく向こう見ずな連中が挑みに行っては黒焦げになっているそうだ」
PRINTL
PRINTFORMW 「……ただどんな魔物なのかは気になるんだよなぁ、誰もはっきりと姿を見たことはないんだと」

@QST_ADD_EVENT_QUEST_42_EVENT_3
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "よく日焼けした漁師の男"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_42_EVENT_3
PRINTFORMW 「岬の方には灯台があるんだが、曾祖父さん聞いた話だとあの辺りは昔は祭祀場だったらしい」
PRINTFORMW 「なんでも昔は歌好きの気の良い神様がいて、秋口になるとあの辺りで祭りをやって
PRINTFORMW 歌や踊りを披露したり捧げ物をしていたそうだ」
PRINTL
PRINTFORMW 「んで神様の方は見返りに盛大に雨を降らせてくれたんだと」
PRINTL
PRINTFORMW 「おっかねぇ魔物がうようよしてる今じゃ考えられん話だね」

@QST_ADD_EVENT_QUEST_42_EVENT_4
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "町外れの教会"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_42_EVENT_4
PRINTFORMW 町外れの教会を訪れると若いシスターが会釈した
PRINTL
PRINTFORML 「この辺りで正教を信仰されている方はあまりおりませんね」
PRINTFORMW 「古来より海神様の信仰が根強い地域ですから」
PRINTL
PRINTFORMW 「ただ最近は古来の海神様とは異なる、異海の神を祀られる方もいらっしゃるようです」
PRINTL
PRINTFORMW 「ただその……あまり良い噂を聞かないというか……あっ、ごめんなさい。失言でした……」

@QST_ADD_EVENT_QUEST_42_EVENT_9
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "大通りの土産屋"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_42_EVENT_9
CALL HO_SHOP_MENU("QUEST_42_土産屋", "大通りの土産屋", "QUEST_42_DIC_SHOP_STOCK_0")

;-------------------------------------------------
;ショップの定義
;関数名は@HO_SHOP_MANAGE_SHOP_{ショップ名称}で定義
;【メモ】
;在庫データをセーブするために専用の広域変数を確保している。
;在庫をセーブに保持する必要がないならDICS_SHOP_STOCK:nに直接販売したいアイテムの情報を書き込めば良い
;-------------------------------------------------
@HO_SHOP_MANAGE_SHOP_QUEST_42_土産屋(DICS_SHOP_STOCK)
#DIMS REF DICS_SHOP_STOCK

QUEST_42_DIC_SHOP_STOCK_0:0 '= @"[CLOTH_ID:11][CLOTH_PART:{衣装部位_下着上}][ITEM_PRICE:7500]"
QUEST_42_DIC_SHOP_STOCK_0:1 '= @"[CLOTH_ID:23][CLOTH_PART:{衣装部位_下着上}][ITEM_PRICE:7500]"
QUEST_42_DIC_SHOP_STOCK_0:2 '= @"[CLOTH_ID:20][CLOTH_PART:{衣装部位_下着下}][ITEM_PRICE:7500]"
QUEST_42_DIC_SHOP_STOCK_0:3 '= @"[CLOTH_ID:26][CLOTH_PART:{衣装部位_下着下}][ITEM_PRICE:7500]"
QUEST_42_DIC_SHOP_STOCK_0:4 '= @"[CLOTH_ID:23][CLOTH_PART:{衣装部位_下着下}][ITEM_PRICE:1000]"
QUEST_42_DIC_SHOP_STOCK_0:5 '= @"[CLOTH_ID:11][CLOTH_PART:{衣装部位_下服}][ITEM_PRICE:5500]"
QUEST_42_DIC_SHOP_STOCK_0:6 '= @"[CLOTH_ID:10][CLOTH_PART:{衣装部位_靴}][ITEM_PRICE:3500]"
QUEST_42_DIC_SHOP_STOCK_0:7 '= @"[ITEM_ID:{ITEM_ローション}][ITEM_PRICE:1600]"
;初回読み込み時のみ在庫として追加
IF QUEST_42_DIC_SHOP_STOCK_0:8 == ""
    {
        QUEST_42_DIC_SHOP_STOCK_0:8 '= @"[WEAPON_ID:6][EQUIP_TYPE:1]"
            + @"[WEAPON_NAME:珊瑚のワンド]"
            + @"[WEAPON_RANK:0][MATERIAL:{WPN_材質_その他}]"
            + @"[ACTIVE_WEAPON_SKILL:WPN_ADD_CAST_SELF,1,]"
            + @"[ITEM_PRICE:50000][SHOP_STOCK:1]"
    }
ENDIF
ARRAYCOPY "QUEST_42_DIC_SHOP_STOCK_0", "DICS_SHOP_STOCK"

repo_name: pervysquirrelz/erafl

license: unknown


code:
;-------------------------------------------------
;EV40 難破船
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_42_EVENT_40
; #region
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_42_EVENT_40
CALL PRINT_TITLE("難破船")
WAIT
PRINTFORMW %QST_LEADER_NAME()%は海岸線に沿って北へと歩みを進めた
PRINTFORMW 入り組んだ浜辺を道なりに進むと岩礁に木造の帆船が乗り上げ、朽ち果てているのを見出した
PRINTFORMW 周囲には木片に布、縄、樽や木箱といった積み荷が散乱している
PRINTL
PRINTFORMW 人の気配は感じられない。様子を見るに座礁してから随分と時間が経っているようだ
PRINTFORMW 漁ってみれば何か発見があるかもしれない
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)

;宝箱
@QST_ADD_EVENT_QUEST_42_EVENT_46
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME

EVENT_SLOT_NAME '= @"宝箱"
;取得済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

;イベント実行内容
@QST_QUEST_42_EVENT_46
#DIMS DYNAMIC _ITEM_LIST
#DIM  DYNAMIC CHEST_TYPE

_ITEM_LIST '= @"WEAPON_ブラウン・ベス,1,"
CALL QST_CEV_LOCKED_CHEST(_ITEM_LIST,125,100)
IF RESULT
    CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
ENDIF

;樽
@QST_ADD_EVENT_QUEST_42_EVENT_47
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME

EVENT_SLOT_NAME '= @"樽"
;取得済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

;イベント実行内容
@QST_QUEST_42_EVENT_47
#DIMS DYNAMIC _ITEM_LIST
#DIM  DYNAMIC CHEST_TYPE

_ITEM_LIST '= @"{ITEM_蒸留酒},5,"
CALL QST_CEV_CHEST(_ITEM_LIST)
IF RESULT
    CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
ENDIF

;沖に浮かぶ宝箱
@QST_ADD_EVENT_QUEST_42_EVENT_48
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME

EVENT_SLOT_NAME '= @"沖に浮かぶ宝箱"
;取得済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

;イベント実行内容
@QST_QUEST_42_EVENT_48
#DIMS DYNAMIC _ITEM_LIST

PRINTFORMW %QST_LEADER_NAME()%は少し沖にぷかぷかと宝箱が浮いているのを見出した
PRINTFORMW 少し泳げば取りに行くことも出来そうだ
PRINTL
PRINTFORMW しかしよく見ると鮫を思わせる背鰭が複数、少し離れたところで回游しているのが見える
PRINTFORMW 背鰭の大きさを見るに恐ろしく大型の個体のようだ。魔物と考えるほうがよいだろう
PRINTL
PRINTFORMW さて……どうしたものか
PRINTL
CALL QST_ROLL_MULTI("身体,歌唱,", "意を決して泳ぐ(身体),声を張り上げておびき出す(歌唱),", "400,250,", 8)
IF RESULT == 1
    SELECTCASE RESULT:2
        CASE 0
            PRINTFORMW %ANAME(RESULT:1)%は意を決して海に飛び込むと、鮫に気付かれる前に宝箱まで泳ぎ切る！
            PRINTFORMW そして素早く宝箱を回収すると、仲間たちの元へと戻った
            GOTO EV_48_GET_TREASURE
        CASE 1
            PRINTFORMW ……おびき出して先に危険を排除してしまえばいい
            PRINTFORMW そう考えた%ANAME(RESULT:1)%は大きく息を吸い込むと、声を張り上げて叫んだ
            PRINTL
            PRINTFORMW そうするやいなや、背鰭は素早く%QST_LEADER_NAME()%の方へターンし、襲いかかってきた
            PRINTFORMW 狙い通りだ。武器を構え、迎撃するとしよう

            ;戦闘イベント初期化
            CALL CC_INIT()
            ;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
            CALL CC_ADDPARTY()
            ;エネミーを防御側で追加
            CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "メガロドン", "メガロドン１")
            CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "エビルシャーク", "エビルシャーク")
            ;戦闘イベント呼出
            CALL CLOSE_COMBAT()
            IF RESULT == 1
                CALL CC_APPLY_CHARA_HP
                CALL CC_GET_LOOT_FROM("DEFENDER")
                PRINTFORMW 目論見通り魔物を誘い出し、殲滅することができた
                PRINTFORMW 脅威を排除した%QST_LEADER_NAME()%は宝箱の確保に成功した！
                CALL QST_CHANGE_ITEM("名誉点,120,")
                GOTO EV_48_GET_TREASURE
            ELSE
                ;汎用敗北イベントへ
                CALL QST_QUEST_42_DEFEAT_EVENT
            ENDIF
    ENDSELECT
ELSEIF !RESULT
    SELECTCASE RESULT:2
        CASE 0
            PRINTFORMW %ANAME(RESULT:1)%が意を決して海に飛び込むやいなや、無数の背鰭が一直線にこちらに向かってきた！
            PRINTFORMW もはや泳ぐどころではない、早く迎撃の体勢を整えねば！
        CASE 1
            PRINTFORMW ……おびき出して先に危険を排除してしまえばいい
            PRINTFORML そう考えたところまでは良かった。しかし%ANAME(RESULT:1)%が声を張り上げるやいなや
            PRINTFORMW 海面から更に複数の背鰭が頭を覗かせ、%QST_LEADER_NAME()%めがけ襲いかかってきた！
            PRINTL
            PRINTFORMW 想定以上に多くの敵を引き寄せてしまったようだ！　兎にも角にも迎え撃たねば！
    ENDSELECT

    ;戦闘イベント初期化
    CALL CC_INIT()
    ;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
    CALL CC_ADDPARTY()
    ;エネミーを防御側で追加
    CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "メガロドン", "メガロドン１")
    CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "メガロドン", "メガロドン２")
    CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "エビルシャーク", "エビルシャーク１")
    CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "エビルシャーク", "エビルシャーク２")
    ;戦闘イベント呼出
    CALL CLOSE_COMBAT()
    IF RESULT == 1
        CALL CC_APPLY_CHARA_HP
        CALL CC_GET_LOOT_FROM("DEFENDER")
        PRINTFORMW 想定通りにはいかなかったが、なんとか魔物の群れを殲滅することができた……
        PRINTFORMW 脅威を排除した%QST_LEADER_NAME()%は宝箱の確保に成功した！
        CALL QST_CHANGE_ITEM("名誉点,80,")
        GOTO EV_48_GET_TREASURE
    ELSE
        ;汎用敗北イベントへ
        CALL QST_QUEST_42_DEFEAT_EVENT
    ENDIF
ELSE
    PRINTFORMW リスクを冒すべきではない、やめておこう
    RETURN
ENDIF

$EV_48_GET_TREASURE
CALL QST_CHANGE_ITEM("RELIC_真珠の首飾り,2,")
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
CALL QST_DEACTIVATE_THIS_EVENT

; #endregion


code:
;-------------------------------------------------
;EV90-120 入り江関連イベント
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_42_EVENT_91
; #region
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME

EVENT_SLOT_NAME '= @"腰掛けるマーメイド"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_42_EVENT_91
#DIM DYNAMIC EVENT_FLAG
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT

;初回
IF EVENT_FLAG == 0
    PRINTFORMW 「あら、この辺りでヒトを見かけるのは珍しいわね」
    PRINTFORMW %QST_LEADER_NAME()%が岩礁伝いに歩みを進めると、不意に誰かが声をかけてきた。
    PRINTFORMW 声の方に目をやると、半人半漁、マーメイドの女が岩礁に腰掛け、ひらひらと手を振っている
    PRINTFORMW 「近頃はどいつもこいつもすっかり腑抜けちゃったのか、ビビってこっち側まで来ないのよ」
    PRINTL
    PRINTFORMW 人魚の女はヒレをひらひらと動かしながら続ける
    PRINTL
    PRINTFORMW 「その点あんた達は結構面白そう。ねぇ、一つ頼まれてくれないかしら？」
    PRINTL
    PRINTFORMW 「最近スキュラが馬鹿みたいに増えちゃって困ってるの。連中繁殖期になると見境なく盛るから……」
    PRINTFORMW 「そこでこの先の連中の巣穴まで行って、ちょちょいとひと暴れしてきて欲しいの」
    PRINTL
    PRINTFORMW 「冒険者なら得意でしょ？　そういうの」
    PRINTFORMW 「勿論報酬は支払うわ。金銀財宝、そういうのも……好きでしょう？　ヒトって」
    PRINTL
    PRINTL
    PRINTFORMW さて、どうしたものだろう。人魚の提案を受けてもいいし、断ってもいい
    PRINTL
    CALL ASK_YN("提案を受ける", "断る")

    IF RESULT
        PRINTFORMW 「あっそ。見る目が無いわねぇ……人魚に恩を売れるチャンスなんてめったに無いわよ、損はさせないのにねぇ……」
        CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
        RETURN
    ENDIF

    PRINTFORMW 「うんうん、見込んだ通りね。この岩礁をそのまま進めば連中が巣穴にしてる洞窟に行き当たるわ」
    PRINTFORMW 「連中、正面の入口はガチガチに固めているから、真正面から突っ込むのは……自殺みたいなものね」
    PRINTL
    PRINTFORML 「ただ連中も気が付いてない抜け道があるの」
    PRINTFORMW 「手前の入り江の潮溜まりの底から、洞窟の中に直結してる抜け道があるからそれを使うといいわ」
    PRINTL
    PRINTFORMW 「じゃあ、よろしく頼むわねー」
    CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 2)
    ;一度提案を拒否
ELSEIF EVENT_FLAG == 1
    PRINTFORMW 「私の提案、受けてくれる気になった？」
    CALL ASK_YN("提案を受ける", "断る")
    IF RESULT
        PRINTFORMW 「あっそ。この通りもう充分濡れてるでしょう？　冷やかしなら間に合ってるわ」
        RETURN
    ENDIF
    PRINTFORMW 「うんうん、見込んだ通りね。この岩礁をそのまま進めば連中が巣穴にしてる洞窟に行き当たるわ」
    PRINTFORMW 「連中、正面の入口はガチガチに固めているから、真正面から突っ込むのは……自殺みたいなものね」

    PRINTFORML 「ただ連中も気が付いてない抜け道があるの」
    PRINTFORMW 「手前の入り江の潮溜まりの底から、洞窟の中に直結してる抜け道があるからそれを使うといいわ」
    PRINTL
    PRINTFORMW 「じゃあ、よろしく頼むわねー」
    CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 2)
    ;提案受諾後
ELSEIF EVENT_FLAG > 1
    PRINTFORMW 「何？　お喋りしたいの？　別にいいけれど」
    CALL QST_QUEST_42_EVENT_TALK_WITH_MERMAID
ENDIF

@QST_QUEST_42_EVENT_TALK_WITH_MERMAID
#DIM EVENT_FLAG
#DIM EVENT_FLAG_SKYLLA
$INPUT_LOOP
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
EVENT_FLAG = RESULT
;スキュラの女王イベントの進捗?
CALL QST_GET_SV_EVENT_FLAG(120)
EVENT_FLAG_SKYLLA = RESULT
PRINTL
{
    CALL ASK_MULTI_JUDGE_HIDE_DEACTIVE(
        "スキュラと話をつけてきた", EVENT_FLAG_SKYLLA == 2 && EVENT_FLAG == 2
        ,"スキュラどもを根絶やしにしてきた", EVENT_FLAG_SKYLLA == 9 && EVENT_FLAG == 2
        ,"マーメイドについて", 1
        ,"この辺りのことについて", 1
        ,"岬の怪物", 1
        ,"用はない", 1)
}
SELECTCASE RESULT
    CASE 0
        PRINTFORML
        PRINTFORMW 「えぇ……それで丸め込まれてきちゃったの？」
        PRINTFORMW 「きちんと言葉が通じるスキュラなんて…ちょっと信じられないけれど、まぁしばらく大人しくしててくれるなら別にいいか……」
        PRINTFORMW 「まぁやることはやってきてくれたみたいだし、はい、これ上げるね」
        CALL QST_CHANGE_ITEM("RELIC_人魚のハープ,1,WEAPON_人魚のトライデント,1,")
        CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 3)
    CASE 1
        PRINTFORML
        PRINTFORMW 「わ、わぁ…根切り……根切りかー……。見かけによらず結構過激なのねぇ……」
        PRINTFORMW 「ちょっと暴れてくれるくらいで良かったのだけど……、まぁ約束は約束だから。はい、これ上げるね」
        CALL QST_CHANGE_ITEM("RELIC_人魚のハープ,1,WEAPON_人魚のトライデント,1,")
        CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 3)
    CASE 2
        PRINTFORML
        PRINTFORMW 「一口にマーメイドといってもそれはもういろんな生き方をしてる氏族がいるから、説明するの難しいんだけどねー」
        PRINTFORML 「ただ私達みたいに歌って、食っちゃ寝して、また歌って暮らしているのもいるし、
        PRINTFORMW 戦争が終わったのにまだ人間の下であくせく武器を振ってる連中もいる」
        PRINTFORMW 「人間の下で働いて一体何が楽しいのやら」
    CASE 3
        PRINTFORML
        PRINTFORMW 「んー、昔に比べれば随分落ち着いたみたいね。あ、これは海の中の話ね」
        PRINTFORMW 「昔は海の中もわんさか魔物がいて、年がら年中戦っていたみたいだけど今はまぁのほほんとしてるわね」
        PRINTFORMW 「ただ、近頃は魚人族が妙な信仰を担ぎ出したり、きな臭いのは相変わらず。やんなっちゃうわね」
    CASE 4
        PRINTFORML
        PRINTFORMW 「岬の方に行くなら気をつけることねー。特に秋の、海が荒れる時期は」
        PRINTFORMW 「あの辺りはものすごーくおっかないヌシ様がいて、機嫌が悪いと雷を落としてくるんだって」
        PRINTFORMW 「お婆様が言うには昔は優しくて、お祭りの歌なんかを聞きにやってきたそうなんだけどね」
        PRINTFORML
        PRINTFORMW 「岬の先で歌うの気持ちよさそうなんだけど、ちょっとおっかなくて無理だねー」
    CASEELSE
        PRINTFORMW 「そう。またねー」
        RETURN
ENDSELECT
GOTO INPUT_LOOP

; #endregion

@QST_ADD_EVENT_QUEST_42_EVENT_100
; #region
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = @"潮溜まり"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_42_EVENT_100
#DIM DYNAMIC EVENT_FLAG
PRINTFORMW 潮の溜まった水溜りがある
PRINTFORMW そこそこの深さがあり、小魚が泳いでいるのを見て取れるが、それ以外に変わった様子はない

@QST_ADD_EVENT_QUEST_42_EVENT_101
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = @"潮溜まり"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_42_EVENT_101
;EV100と共通
CALL QST_QUEST_42_EVENT_100

@QST_ADD_EVENT_QUEST_42_EVENT_102
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = @"潮溜まり"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%

@QST_QUEST_42_EVENT_102
#DIM DYNAMIC EVENT_FLAG
#DIM DYNAMIC EVENT_FLAG_SKYLLA
CALL QST_GET_SV_EVENT_FLAG(91)
EVENT_FLAG = RESULT
CALL QST_GET_SV_EVENT_FLAG(120)
EVENT_FLAG_SKYLLA = RESULT

;人魚の依頼を受けている場合分岐
IF EVENT_FLAG == 2
    PRINTFORMW 人魚に言われた通り潮溜まりをよく観察してみると海藻が揺らぐ水底に、更に奥へ続く穴が空いているのが見て取れた
    PRINTFORMW ここからスキュラの巣穴に直接向かうことができるそうだ……
    PRINTL
    PRINTFORMW ただし、自ずから海魔どもの巣窟の真っ只中に飛び込もうというのなら相応の準備は済ませておくべきであろう
    PRINTFORMW おそらく出口に達した瞬間から激戦は避けられないはずだ
    PRINTL
    CALL ASK_YN("潜って進む", "やめておく")
    IF RESULT
        RETURN
    ENDIF
    PRINTFORMW %QST_LEADER_NAME()%は意を決して潮溜まりに飛び込んだ！
    PRINTL
    PRINTFORMW 海藻を払い除け、穴の先を目指す。洞穴は途中で真横に向かって屈曲しており、
    PRINTFORMW 程なくして呼吸可能な横穴へと出た
    PRINTL
    PRINTFORMW この先がスキュラどもの巣窟らしい。先へと進むとしよう
    PRINTL
    CALL HO_MOVE_MASTER(12)
ELSE
    ;EV100と共通
    CALL QST_QUEST_42_EVENT_100
ENDIF

@QST_ADD_EVENT_QUEST_42_EVENT_110
; #region
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = ""
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_42_EVENT_110
;スキュラの女王を倒しているならこのイベントは解決済みとして終了
CALL QST_GET_SV_EVENT_FLAG(120)
IF RESULT > 1
    CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
    RETURN
ENDIF

CALL PRINT_TITLE("スキュラの巣窟")
WAIT
PRINTFORMW %QST_LEADER_NAME()%は複雑に入り組む入り江の奥までやってきた
PRINTFORMW 入り江の奥には大きな洞窟がぽっかりと口を開けているのが見て取れる
PRINTFORMW 洞窟の手前の水面に目をやると、無数の人型の影が水面下を行き来している
PRINTL
PRINTFORMW 上半身は裸体の女、下半身からは複数の触手を生やし、触手を器用に繰って水底を疾駆している
PRINTFORMW その魔物はスキュラとして知られている。おそらく歩哨の役割を果たしているのだろう
PRINTFORML
PRINTFORMW もしこのまま先へ進めば、スキュラの群れと真正面からやり合うことになる
PRINTFORMW 連戦は避けられないだろう
PRINTL
PRINTFORMW さて、どうしたものか……
PRINTL
CALL ASK_YN("一度引き返す", "強行突破する")
IF !RESULT
    PRINTFORMW 真正面から突っ込むのは無謀と考えた%QST_LEADER_NAME()%は何か策が無いか考えてみることにした
    ;先にはスキュラを討伐しない限り移動できないようにする
    ;・1マス戻ってから12.海魔の巣窟を非表示にする
    CALL HO_MOVE_MASTER(10)
    CALL QST_SET_FOG_OF_WAR(12)
    RETURN
ENDIF
PRINTFORMW ――小細工は無用、血路を切り拓くとしよう！
PRINTFORMW あなたたちは武器を取り、海魔の群れへと襲いかかった！

;戦闘イベント初期化
CALL CC_INIT()
;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
CALL CC_ADDPARTY()
;エネミーを防御側で追加
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "スキュレー・センチネル", "スキュレー・センチネル１")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "スキュレー・センチネル", "スキュレー・センチネル２")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "エビルシャーク", "エビルシャーク")
;戦闘イベント呼出
CALL CLOSE_COMBAT()
IF RESULT == 1
    CALL CC_APPLY_CHARA_HP
    CALL CC_GET_LOOT_FROM("DEFENDER")
    PRINTFORMW %QST_LEADER_NAME()%は押し寄せるスキュラの第一陣を打ち倒した
    PRINTFORMW ――しかし続けざまに第二陣が殺到する！
ELSE
    ;汎用敗北イベントへ
    CALL QST_QUEST_42_DEFEAT_BY_SKYLLA
    RETURN
ENDIF

;戦闘イベント初期化
CALL CC_INIT()
;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
CALL CC_ADDPARTY()
;エネミーを防御側で追加
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "スキュレー・テンタクラー", "スキュレー・テンタクラー１")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "スキュレー・テンタクラー", "スキュレー・テンタクラー２")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "スキュレー・センチネル", "スキュレー・センチネル")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "メガロドン", "メガロドン")
;戦闘イベント呼出
CALL CLOSE_COMBAT()
IF RESULT == 1
    CALL CC_APPLY_CHARA_HP
    CALL CC_GET_LOOT_FROM("DEFENDER")
    PRINTFORMW ……なんとか殺到するスキュラの群れを打ち払うことが出来た
    PRINTFORMW これで先に進むことができそうだ
ELSE
    ;汎用敗北イベントへ
    CALL QST_QUEST_42_DEFEAT_BY_SKYLLA
    RETURN
ENDIF

;イベントクリアフラグON
CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 1)
;戦闘が再発生しないようにイベントを非アクティブにする
CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
;探索済に設定する
CALL QST_ADD_LIST_EXPLORED_ROOM(11, 1)

; #endregion

@QST_ADD_EVENT_QUEST_42_EVENT_120
; #region
#DIMS EVENT_TRIGGER = "ENTER_ROOM"
#DIMS EVENT_TYPE = "ONCE"
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:2 = %EVENT_TYPE%
; ゲーム中一度でも解決済みの場合は非アクティブに
CALL QST_GET_SV_EVENT_FLAG(QST_NOWEV)
IF RESULT
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF

@QST_QUEST_42_EVENT_120
#DIM DYNAMIC CHARA_ID
CALL FL_PLAYBGM("Ambient_dark_2.mp3")
CALL PRINT_TITLE("女王の閨")
WAIT

PRINTFORMW %QST_LEADER_NAME()%は洞窟の深部へと歩みを進めた
PRINTFORMW 内部は石柱の林立する鍾乳洞になっており、奥へ進むほどに洞窟は空間的な広さを増していく
PRINTFORMW 洞窟の内部は生温かく、潮の生臭さに包まれており快適とは言い難い
PRINTL
PRINTFORMW 加えて、特にあなたを苛んだのは洞窟を満たす独特の甘い香りであった
PRINTFORMW 酷く濃い香水のような香りが空間を満たしており、その臭気は進めば進むほどに強さを増していった
PRINTFORMW
PRINTFORMW ……これは繁殖期のスキュラの放つフェロモンだ。増す香りは連中の寝蔵が近いことを示している
PRINTFORMW 臭気をたどり%QST_LEADER_NAME()%は最深部へと至った
PRINTL
PRINTFORMW そこで%QST_LEADER_NAME()%が目にしたのは……膨大な肉のうねりであった
PRINTFORMW 無数の雄役と雌役のスキュラが互いに絡みつき、絡み合い交接に耽っている
PRINTFORMW 互いに無数の触手を絡めあい、生殖器の激しい注挿を繰り返す
PRINTFORMW 個体が絶頂に至る度に、美女の姿をとる上半身が歓喜に悶え、快楽の叫びを上げた
PRINTL
PRINTL
CALL COLOR_PRINTW("「……逢瀬の最中に押し入るとは、とんだ無粋な連中もいたものよ」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
PRINTL
PRINTFORMW 不意によく響く、刺すような声が頭上から投げかけられる
PRINTFORMW 肉欲のうねりの頂きに一体のスキュラが鎮座している
PRINTFORMW 無数の宝飾を身にまとい、冷たく蔑むような瞳で%QST_LEADER_NAME()%を見下ろしている
PRINTL
CALL COLOR_PRINTW("「何ゆえ我らの快楽と安寧を脅かす？　目当ては金か？　名誉か？」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
PRINTL
PRINTFORMW スキュラの主だろうか。流暢に人語を操り続ける
CALL COLOR_PRINTW("「まぁどちらでもよい。妾の閨を汚した者共を活かしておく道理などないのだから」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
CALL COLOR_PRINTW("「……八つ裂きにして喰ろうてやろうか、余興代わりに嬲り、慰み者にしてやろうか」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
PRINTL
PRINTFORMW スキュラの主が一度命を下すと、姦淫に耽っていた群体がゆっくりと身体を起す
PRINTFORMW しゅるしゅると触手が伸び、鎌首をもたげる
PRINTL
PRINTL
PRINTFORMW 状況を打破するには武器を振るうしかないようだ！

;戦闘イベント初期化
CALL CC_INIT()
;クエストのパーティメンバー全員を攻撃側として戦闘メンバーに追加
CALL CC_ADDPARTY()
;エネミーを防御側で追加
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "QUEST_42_『デミスキュレー・クイーン』", "『デミスキュレー・クイーン』")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "スキュレー・エヴゾネス", "スキュレー・エヴゾネス１")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "スキュレー・ソーサレス", "スキュレー・ソーサレス１")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "スキュレー・ソーサレス", "スキュレー・ソーサレス２")
CALL CC_ADDMOB_FROM_TABLE("DEFENDER", "スキュレー・テンタクラー", "スキュレー・テンタクラー")
;戦闘イベント呼出
CALL CLOSE_COMBAT("[BGM:Boss_3.ogg]")
IF RESULT == 1
    CALL FL_PLAYSE("Fanfare_0.ogg")
    CALL FL_PLAYBGM("Nostalgic_2.ogg")
    CALL CC_APPLY_CHARA_HP
    CALL CC_GET_LOOT_FROM("DEFENDER")
    PRINTL
    PRINTFORMW 無数に迫りくる触手を打ち払い、引き倒し、あなたたちはスキュラの主へと距離を詰める
    PRINTFORMW そしてついに刃が首魁を捉えた！
    PRINTL
    CALL COLOR_PRINTW("「ま、待て、貴様の強さは良くわかった。ここは一つ、取引に応じる気はないか」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    PRINTL
    IF QST_PARTYNUM > 1
        PRINTFORMW ――魔物の言葉は偽りであり耳を貸すべきではない、と仲間が諌める
    ELSE
        PRINTFORMW ――魔物の言葉は偽りであり耳を貸すべきではない、と心の声が諌める
    ENDIF
    PRINTL
    CALL COLOR_PRINTW("「他の眷属は兎も角、妾は全く話の通じぬ相手というわけではない」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    PRINTL
    CALL COLOR_PRINTW("「妾とその一族には幾らかヒトの血が流れている。幸か不幸かスキュラの番となったヒトの女のな」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    CALL COLOR_PRINTW("「他の個体と異なり妾は幾分ヒトの血が濃く出たらしい。その血故にヒトにも魔物にも交わることができず、この狭い入り江の外では生きていくことも叶わぬ」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    CALL COLOR_PRINTW("「故にこうしてひっそりと悦楽に耽って暮らしているのだ」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    PRINTL
    CALL COLOR_PRINTW("「我らはそれ以上を望まぬ。この小さな領地が侵されぬ限り他者と対峙するつもりもない。我らの方からはな」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    PRINTL
    PRINTL
    CALL COLOR_PRINTL("「さて、もし貴様らが取引に応じるなら、今後貴様らを\"賓客\"として遇し、我らの体液を幾らかわけてやってもいい」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    CALL COLOR_PRINTW("「貴様らがそれを大層重宝していることは知っておる」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    PRINTL
    PRINTFORML ……彼女の言う通りスキュラのフェロモンから精製される香水は「メイガスの夜」と呼ばれ非常に高価に取引されている
    PRINTFORMW 原材料である体液もまた然りだ
    PRINTL
    PRINTFORMW さて、このスキュラの言葉にはもしかしたら一考の余地はあるかもしれない
    PRINTFORMW 無論魔物の戯言と判断し、斬り伏せるのも自然な選択だ
    PRINTL
    PRINTFORMW 判断は%QST_LEADER_NAME()%に委ねれている。どうしたものだろうか？
    PRINTL
    CALL ASK_YN("取引に応じる", "根切りにする")
    PRINTL
    IF RESULT
        CALL COLOR_PRINTW("「ま、待っ――」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
        PRINTFORML 甘言に惑わされるべきではない
        PRINTFORMW そう判断した%QST_LEADER_NAME()%はスキュラの主が二の句を告げる前に斬り捨て、生き残りを淡々と鏖殺した
        PRINTFORML
        PRINTFORMW この地から海魔の命脈の一つを完全に絶った。人の世を安寧へと一歩近づけたと言っていいだろう
        CALL QST_CHANGE_ITEM("名誉点,100,")
        ;女王死亡処理
        CALL QST_QUEST_42_SCYLLA_QUEEN_DEAD()
        ;イベントクリアフラグON (9.根切りエンド)
        CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 9)
        ;探索済に設定する
        CALL QST_ADD_LIST_EXPLORED_ROOM(12, 1)
    ELSE
        PRINTFORMW %QST_LEADER_NAME()%が取引に応じると告げるとスキュラの主はほっと胸を撫で下ろす
        PRINTL
        CALL COLOR_PRINTW("「話の分かる連中で一安心した……。ほれ、約束の品だ」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
        PRINTL
        PRINTFORMW スキュラの主の触手から、非常に濃厚な甘い香りのする粘性の体液が垂れ、小瓶を満たしていく……
        CALL QST_CHANGE_ITEM(@"{ITEM_スキュラのフェロモン},3,淫堕点,100,")
        PRINTL
        CALL COLOR_PRINTW("「ところで……妾は一族のさらなる安寧のため強い種を求めておる」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
        CALL COLOR_PRINTW("「もし貴様が望むなら……貴様の種と引き換えにこれからも妾の体液を分けてやろうぞ」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
        PRINTL
        PRINTFORMW  触手をねっとりとのたくらせ、女王は蠱惑的な笑みを浮かべる
        CALL COLOR_PRINTW("「妾を打ち負かした貴様であれば申し分はない」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
        PRINTL
        CALL COLOR_PRINTW("「もっとも……今は出涸らしよ……。間を置いてまた来るがいい。人の身では味わえぬ無上の快楽をくれてやろうぞ❤」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
        PRINTL
        PRINTFORMW そう言い残し、スキュラの女王は洞窟の奥へ引いていった……
        ;イベントクリアフラグON (2.和解エンド)
        CALL QST_SET_SV_EVENT_FLAG(QST_NOWEV, 2)
        ;探索済に設定する
        CALL QST_ADD_LIST_EXPLORED_ROOM(12, 1)
    ENDIF
ELSE
    CALL CC_APPLY_CHARA_HP
    CALL FL_PLAYBGM("Ambient_dark_2.mp3")
    PRINTFORMW スキュラの強靭な触手が%QST_LEADER_NAME()%の武器を弾き飛ばす！
    PRINTFORMW 強力な触手の一撃と魔法を駆使して戦うスキュラの戦術に%QST_LEADER_NAME()%は次第に追い詰められていく……
    PRINTL
    PRINTFORMW そうして隙を見せたところ、とうとう足元から忍び寄ってきた触手に絡め取られてしまった
    PRINTL
    PRINTL
    CALL COLOR_PRINTW("「くくっ、なんと他愛ない。所詮ヒトの身で到れる境地とはその程度のものよ」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    CALL COLOR_PRINTW("「……しかし敢闘は褒めてやろう……。喰らうのも轢き潰すのも取り止めてやる。その代わり……」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    PRINTL
    PRINTFORMW スキュラの女王は%QST_LEADER_NAME(1)%を捉える触手を手繰り寄せるとその頬を舌でねっとりと舐める
    PRINTL
    CALL COLOR_PRINTW("「犯して、犯して、犯し抜いて、精を一滴残らず絞り出して、その血脈を妾の眷属に加えてやるとしよう」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    CALL COLOR_PRINTW("「良質な外の血を欲しておったところよ。貴様なら丁度よい」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    CHARA_ID = QST_LEADER_ID()
    IF !HAS_PENIS(CHARA_ID)
        PRINTL
        CALL COLOR_PRINTW("「何？　自分は女だと？　心配は無用よ、ほれ」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
        PRINTL
        PRINTFORMW スキュラの触手が%QST_LEADER_NAME(1)%の口に強引にねじ込まれる。そして喉奥に何かねっとりした液体を流し込んでいく
        PRINTFORMW 苦しさのあまり声にならない声を上げてもがく様を、スキュラの主は恍惚とした笑みを浮かべて眺めている
        PRINTL
        PRINTFORMW しばらくして解放された%QST_LEADER_NAME(1)%は自身の下腹部が猛烈に疼くことに気がつく
        PRINTL
        PRINTFORMW 陰部が痺れ、疼き、はち切れんばかりの劣情がこみ上げる……
        CALL COLOR_PRINTW(@"―――― %ANAME(QST_LEADER_ID())%は一時的にふたなりにされてしまった……！", カラー_ピンク)
        CALL TO_FUTANARI(QST_LEADER_ID())
        CALL SET_PENISSIZE(QST_LEADER_ID(), 1)
        CFLAG:QST_LEADER_ID():一時的ふたなり = 1
        PRINTL
        CALL COLOR_PRINTW("「ほれ、立派なモノがあるではないか。では……」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
        PRINTL
        PRINTFORMW スキュラの群れが粘液を帯びた触手を一斉に展開する……
        PRINTL
        CALL COLOR_PRINTW("「客人も交えて宴の続きを始めるとしよう」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    ENDIF
    ;汎用敗北イベントへ
    CALL QST_QUEST_42_DEFEAT_TRAIN_BY_SKYLLA_2
ENDIF

;-------------------------------------------------
;スキュラ連戦で敗北
;-------------------------------------------------
@QST_QUEST_42_DEFEAT_BY_SKYLLA
#DIM CHARA_ID
#DIM CONST 派生イベントID = 125

CALL FL_PLAYBGM("Ambient_dark_2.mp3")
PRINTFORMW ――多勢に無勢だ！
PRINTL
PRINTFORMW 倒しても倒しても押し寄せてくるスキュラの群れを前に%QST_LEADER_NAME()%は一歩、また一歩と追い詰められていった
PRINTL
PRINTFORMW 絶え間なく繰り出される触手の攻撃をいなしきれず、ついに%QST_LEADER_NAME(1)%は痛打を浴びてしまう
PRINTFORMW ぐらりと世界がひっくり返り、その場に膝をつく……
PRINTL
PRINTFORMW 倒れ込みかけた%QST_LEADER_NAME(1)%の身体をスキュラの触手が絡め取り、自由を奪った
PRINTL
PRINTFORMW スキュラの群れは%QST_LEADER_NAME()%を潮溜まりに寝かせると甲高い声を上げた
PRINTFORMW 彼女らの言語なのだろうか……？　歓喜の叫びのようであるが聞き取ることは叶わない
PRINTL
PRINTFORMW ……歓喜の叫びとともにスキュラ達は触手をしゅるしゅると広げ、%QST_LEADER_NAME()%の身体に覆いかぶさる
PRINTFORMW 眼前に寄せられた造り物の如く整った美麗な女の面が、嗜虐の笑みで酷く歪んだ……

FLAG:ウフフフラグ = 1
FLAG:クエスト_刻限による終了無効 = 1
FLAG:調教モード = 調教_慰安
TFLAG:調教ターン数 = 24

;スキップ許可
FLAG:調教スキップ可 = 1

;主導除外フラグが立ったキャラクターは慰安調教で常にターゲットに回る。
;拘束中など自主的に行動させたくないシチュエーションでこのフラグを立てる。
CVARSET CFLAG, GETNUM(CFLAG, "調教参加フラグ") , 0
FOR LOCAL, 0, QST_PARTYNUM
    CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
    ;調教コマンドのプレイヤーにならなくなる。一方的に嬲られるシチュ用
    CFLAG:CHARA_ID:主導除外 = 1
    ;良感情を極めて得にくくなるフラグ
    CFLAG:CHARA_ID:凌辱被害者フラグ = 1
    CFLAG:CHARA_ID:調教参加フラグ = 1
NEXT

;慰安モブを追加
;・スキュラ×2、触手×2、シースラッグ×3
CALL QST_QUEST_42_SET_TRAIN_MOB(0, 2, 2, 3)

CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(派生イベントID)

RETURN

;-------------------------------------------------
;スキュラ敗北調教2
;-------------------------------------------------
@QST_QUEST_42_DEFEAT_TRAIN_BY_SKYLLA_2
#DIM CHARA_ID
#DIM CONST 派生イベントID = 125

FLAG:ウフフフラグ = 1
;FLAG:クエスト_刻限による終了無効 = 1
FLAG:調教モード = 調教_慰安
TFLAG:調教ターン数 = 24

;スキップ許可
FLAG:調教スキップ可 = 1

CVARSET CFLAG, GETNUM(CFLAG, "調教参加フラグ") , 0
FOR LOCAL, 0, QST_PARTYNUM
    CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
    ;調教コマンドのプレイヤーにならなくなる。一方的に嬲られるシチュ用
    CFLAG:CHARA_ID:主導除外 = 1
    ;良感情を極めて得にくくなるフラグ
    CFLAG:CHARA_ID:凌辱被害者フラグ = 1
    CFLAG:CHARA_ID:調教参加フラグ = 1
NEXT

;慰安モブを追加
;・女王、スキュラ×2、触手×2
CALL QST_QUEST_42_SET_TRAIN_MOB(1, 2, 2, 0)

CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(派生イベントID)

RETURN

;EV125 敗北調教後イベント
;@QST_QUEST_6_DEFEAT_TRAIN_1終了後に呼ばれる
@QST_QUEST_42_EVENT_125
#DIM DYNAMIC CHARA_ID

;一時ふたなり解除
FOR LOCAL, 0, QST_PARTYNUM
    CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
    IF CFLAG:CHARA_ID:一時的ふたなり
        CALL TO_FUTANARI(CHARA_ID)
        CFLAG:CHARA_ID:一時的ふたなり = 0
    ENDIF
NEXT

PRINTL
PRINTL
PRINTFORMW 触手によって散々に犯され絞られ、うめき声の一滴さえ枯れ果てた%QST_LEADER_NAME()%の姿にスキュラは満足げな笑みを浮かべている
PRINTFORMW 眼の前には自身の武器が転がっているのが見える。なんとか手にし、反撃しなければ……
PRINTL
PRINTFORMW だが消耗しきった身体はもはや言うことを聞かない。次第に視界が霞み、暗転していく
PRINTL
PRINTFORMW このままでは……
PRINTFORMW ……………………
PRINTFORMW ………………
PRINTFORMW …………
PRINTL

;汎用敗北イベント2へ
CALL QST_QUEST_42_DEFEAT_EVENT_2

;-------------------------------------------------
; 攻略後イベント
;-------------------------------------------------
@QST_ADD_EVENT_QUEST_42_EVENT_121
; #region
#DIMS EVENT_TRIGGER = "EVENT_SLOT"
#DIMS EVENT_SLOT_NAME = "スキュラの女王"
; このイベントは初期状態で非アクティブ化
; スキュラの女王再訪時にアクティブに
CALL QST_GET_SV_EVENT_FLAG(120)
IF RESULT != 2
    CALL QST_DEACTIVATE_EVENT(QST_NOWEV)
ENDIF
RESULTS:0 = %EVENT_TRIGGER%
RESULTS:1 = %EVENT_SLOT_NAME%
RETURN

@QST_QUEST_42_EVENT_121
#DIM EVENT_FLAG
#DIM CHARA_ID
#DIM SEDUCTIVE
#DIM CONST 派生イベントID = 126
#DIM EXISTS_SELECTED_CHARA

CALL COLOR_PRINTL("「ふふっ、来たのか。望み通り体液はくれてやってもよい」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
CALL COLOR_PRINTW("「ただし引き換えに貴様の種をいただこうか……❤」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
PRINTL
EXISTS_SELECTED_CHARA = QST_QUEST_42_EVENT_121_EXISTS_SELECTED_CHARA()
CALL ASK_MULTI_JUDGE_ANNOTATION_P("身を捧げる", \@EXISTS_SELECTED_CHARA ? 　 # ※ 参加可能なキャラがいません (主人公 または 要:誘蛾の色香Lv2以上)\@, EXISTS_SELECTED_CHARA, "遠慮する", "", 1)
IF RESULT
    CALL COLOR_PRINTW("「なんだ、つまらん。人の身では到底味わえぬ悦楽をくれてやろうというのに」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    RETURN
ENDIF
CALL COLOR_PRINTW("「そう、それでいい。さぁ、身を委ねよ。溶けるような快楽で身も心も満たしてやろう」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
PRINTL
PRINTL
{
    CALL QST_SELECT_CHARA_MULTI(
        "[MESSAGE:身体を捧げるキャラクターを選んでください (主人公 または 要:誘蛾の色香Lv2以上)]"
        +"[SELECTOR:{(CHARA_ID == MASTER) || HAS_SKILL(CHARA_ID, \"SEDUCTION\", スキル_大分類_性技) > 1}]"
        )
}
;キャラが選ばれなかった
IF !SELECTED_CHARA_NUM
    CALL COLOR_PRINTW("「なんだ、つまらん。人の身では到底味わえぬ悦楽をくれてやろうというのに」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    RETURN
ENDIF
PRINTL
CALL COLOR_PRINTW("「では……いただくとしようか❤」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
CHARA_ID = SELECTED_CHARA:0
IF !HAS_PENIS(CHARA_ID)
    PRINTL
    PRINTFORMW  スキュラの女王は%ANAME(CHARA_ID)%に舐めるような目線を送り、値踏みする
    CALL COLOR_PRINTW("「……ところで、貴様がもし牡の快楽に興味があるなら一時の間\"茎\"を授けてやることも出来る」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    CALL COLOR_PRINTW("「脳を焦がすような牡の射精の快楽に浸りたいなら頷くといい❤」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    PRINTL
    CALL COLOR_PRINTW("「なぁに、案ずるな。時が経てばもとに戻る」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    PRINTL
    PRINTFORML %ANAME(CHARA_ID)%は女王の問に……
    CALL ASK_YN("うなずく", "首を横にふる")
    IF RESULT
        CALL COLOR_PRINTW("「そうか……。ではたっぷり、雌の快楽を授けてやるとしよう」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    ELSE
        PRINTFORMW %ANAME(CHARA_ID)%が頷くとスキュラの女王は満足気な笑みを浮かべ、触手を%ANAME(CHARA_ID)%の口にするりと忍びこませる、
        PRINTFORMW そして喉奥にどろりとした液体を流し込んでいく
        PRINTL
        PRINTFORMW %ANAME(CHARA_ID)%の陰部が甘く痺れ、疼き、はち切れんばかりの劣情に身を悶えさせる……
        PRINTL
        CALL COLOR_PRINTW(@"―――― %ANAME(CHARA_ID)%は一時的にふたなりにされてしまった……！", カラー_ピンク)
        CALL TO_FUTANARI(CHARA_ID)
        CALL SET_PENISSIZE(CHARA_ID, 1)
        CFLAG:CHARA_ID:一時的ふたなり = 1
        PRINTL
        CALL COLOR_PRINTW("「ふふっ❤　見事なモノではないか。ではたっぷり扱き上げて、絞り尽くして、牡の射精の快楽をその身体に刻みつけてやろう……❤」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
    ENDIF
ENDIF

CALL FL_PLAYBGM("Train_0.ogg")

;調教イベント
FLAG:ウフフフラグ = 1
FLAG:クエスト_刻限による終了無効 = 1
FLAG:調教モード = 調教_慰安
TFLAG:奉仕モード = 1
TFLAG:奉仕カテゴリ = 奉仕_スキュラが奉仕
TFLAG:調教ターン数 = 32

;スキップ許可
FLAG:調教スキップ可 = 1

CVARSET CFLAG, GETNUM(CFLAG, "調教参加フラグ") , 0
FOR LOCAL, 0, SELECTED_CHARA_NUM
    CHARA_ID = SELECTED_CHARA:LOCAL
    ;調教コマンドのプレイヤーにならなくなる。一方的に嬲られるシチュ用
    CFLAG:CHARA_ID:主導除外 = 1
    CFLAG:CHARA_ID:調教参加フラグ = 1
    TCVAR:CHARA_ID:被奉仕者フラグ = 1
NEXT

;慰安モブを追加
;・女王、スキュラ×2、触手×2, 和姦
CALL QST_QUEST_42_SET_TRAIN_MOB(1, 2, 2, 0, 1)

CALL QST_ENQUEUE_TRAIN_EXIT_EVENT(派生イベントID)

@QST_QUEST_42_EVENT_121_EXISTS_SELECTED_CHARA()
#FUNCTION
#DIM L_CNT
#DIM CHARA_ID

FOR L_CNT, 0, QST_PARTYNUM
    CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
    SIF (CHARA_ID == MASTER) || HAS_SKILL(CHARA_ID, "SEDUCTION", スキル_大分類_性技) > 1
        RETURNF 1
NEXT
RETURNF 0

;EV126 調教後イベント
;@EV121終了後に呼ばれる
@QST_QUEST_42_EVENT_126
#DIM DYNAMIC CHARA_ID

;一時ふたなり解除
FOR LOCAL, 0, QST_PARTYNUM
    CHARA_ID = TOINT(LIST_GET(_LIST_QST_PARTY, LOCAL))
    IF CFLAG:CHARA_ID:一時的ふたなり
        CALL TO_FUTANARI(CHARA_ID)
        CFLAG:CHARA_ID:一時的ふたなり = 0
    ENDIF
NEXT
CHARA_ID = SELECTED_CHARA:0
PRINTFORMW
PRINTFORMW 触手と両性具有の性器の双方を駆使し、姦淫の限りを尽くしたスキュラの女王は恍惚の表情を浮かべ、上体を仰け反らせる
CALL COLOR_PRINTW("「あぁ……なんと素晴らしい……。この焼けるような快楽❤　普段のまぐわいでは得られぬものよ」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
PRINTL
PRINTFORMW 度重なる絶頂に息絶え絶えになった%ANAME(CHARA_ID)%の身体を、スキュラの女王は触手で優しく絡め取り、
PRINTFORMW 頭を豊満な乳房に埋めさせる
PRINTL
CALL COLOR_PRINTW("「約束通り妾を満足させた褒美をくれてやろう……❤。貴様が望むなら何度でも来るといい……」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
CALL COLOR_PRINTW("「ヒトの情婦では味わえぬ快楽で……思考が溶け果てるまで尽くしてやろう……❤」", QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR)
PRINTL
PRINTFORMW そう言い残してスキュラの女王は洞窟の奥へと引っ込んでいった……
CALL QST_CHANGE_ITEM(@"{ITEM_スキュラのフェロモン},3,淫堕点,100,")
CALL QST_DEACTIVATE_EVENT(121)
;BGMを戻す
CALL FL_STOPBGM()
CALL FL_PLAYBGM("Nostalgic_2.ogg")
; #endregion

;-------------------------------------------------
; 調教用モブの生成
;-------------------------------------------------
@QST_QUEST_42_SET_TRAIN_MOB(人数_スキュラの女王, 人数_スキュラ, 人数_触手, 人数_シースラッグ, 和姦フラグ = 0)
#DIM 人数_スキュラの女王
#DIM 人数_スキュラ
#DIM 人数_触手
#DIM 人数_シースラッグ
#DIM 和姦フラグ
#DIM DYNAMIC L_CNT
#DIM DYNAMIC CHARA_ID

;スキュラの女王
;・ユニークキャラなので何人指定しようと最大1人
IF 人数_スキュラの女王
    ;キャラIDの取得(キャラ未生成なら生成)
    CALL QST_QUEST_42_GET_CHARA_ID_SCYLLA_QUEEN()
    CHARA_ID = RESULT

    ;慰安参加用のフラグ設定
    CFLAG:CHARA_ID:調教参加フラグ = 1
    CFLAG:CHARA_ID:慰安参加者 = 1
    CFLAG:CHARA_ID:強制友好化 = 1
    CFLAG:CHARA_ID:慰安参加者非対象化 = 1
    IF 和姦フラグ
        TCVAR:CHARA_ID:奉仕者フラグ = 1
    ENDIF
ENDIF

;スキュラ
FOR L_CNT, 0, 人数_スキュラ
    CALL QST_TRAIN_MOB_PRESET("スキュラ", "スキュラ", 1)
    CHARA_ID = RESULT
    IF 和姦フラグ
        TCVAR:CHARA_ID:奉仕者フラグ = 1
    ENDIF
NEXT

;触手
FOR L_CNT, 0, 人数_触手
    CALL QST_TRAIN_MOB_PRESET(SPLIT_RAND("触手/ブラシ触手/ブラシ触手",1), SPLIT_RAND("ぬめる触手/這う触手",1))
    CHARA_ID = RESULT
    ;一度MPLY,MTARをクリア
    CALL CLEAR_ACTOR()
    ;触手召喚を継続状態でセット
    CALL ADD_MPLY(CHARA_ID)
    CALL SET_EQUIP(GETNUM(TRAINNAME, "触手召喚"), 1)
    CALL DEL_MPLY(CHARA_ID)
    TALENT:CHARA_ID:性別 = 2
    ABL:CHARA_ID:レズ = 10
    ;和姦フラグがある場合は産卵や開発をしない
    IF 和姦フラグ
        TCVAR:CHARA_ID:奉仕者フラグ = 1
        COM_FORBIDDEN:CHARA_ID:T_N("触手産卵") = 1
        COM_FORBIDDEN:CHARA_ID:T_N("触手Ａ産卵") = 1
        COM_FORBIDDEN:CHARA_ID:T_N("触手Ｕ産卵") = 1
        COM_FORBIDDEN:CHARA_ID:T_N("触手乳腺開発") = 1
        COM_FORBIDDEN:CHARA_ID:T_N("触手尿道挿入") = 1
    ENDIF
NEXT

;シースラッグ
FOR L_CNT, 0, 人数_シースラッグ
    CALL QST_TRAIN_MOB_PRESET("蟲", "シースラッグ", RAND:2)
    CHARA_ID = RESULT
    VARSET COM_FORBIDDEN:CHARA_ID:0, 1
    COM_FORBIDDEN:CHARA_ID:T_N("クンニ") = 0
    COM_FORBIDDEN:CHARA_ID:T_N("フェラチオ") = 0
    COM_FORBIDDEN:CHARA_ID:T_N("アナル舐め") = 0
    COM_FORBIDDEN:CHARA_ID:T_N("乳首吸い") = 0
    COM_FORBIDDEN:CHARA_ID:T_N("キス") = 0
    COM_FORBIDDEN:CHARA_ID:T_N("貝合わせ") = 0
    TALENT:CHARA_ID:両刀 = 1
NEXT

;-------------------------------------------------
; 「スキュラの女王」のキャラID取得/生成
;-------------------------------------------------
@QST_QUEST_42_GET_CHARA_ID_SCYLLA_QUEEN()
#DIM DYNAMIC CHARA_ID
#DIM DYNAMIC 種付数
#DIM DYNAMIC 出産数

;生成済みの場合はIDを返す
CHARA_ID = NAME_TO_CHARA("QUEST_42_スキュラの女王")
SIF CHARA_ID != -1
    RETURN CHARA_ID

;未生成の場合はキャラ生成
;ベースはスキュラ(慰安モブ)
CALL QST_TRAIN_MOB_PRESET("スキュラ", "スキュラの女王", 1)
CHARA_ID = RESULT

;ユニークキャラとして生成するのでCSTR:識別名を登録
;・被らないと思うけど一応クエスト番号を付与
CSTR:CHARA_ID:識別名 '= "QUEST_42_スキュラの女王"

;衣装設定
;ティアラを追加
FACE_PARTS:CHARA_ID:IMG_帽子 '= "CC1_帽子_ティアラ_1"
CALL SET_CLOTH_BY_NAME(CHARA_ID, 衣装部位_頭, "ティアラ")
;全裸にしたらポートレートのティアラが表示されなくなったので適当に着せる
CALL SET_CLOTH_BY_NAME(CHARA_ID, 衣装部位_下着上, "ビキニトップ")
CALL SET_CLOTH_BY_NAME(CHARA_ID, 衣装部位_下着下, "ビキニボトム")
;調教用衣装にコピー
調教用衣装:CHARA_ID:衣装部位_頭     = CLOTH:CHARA_ID:衣装部位_頭
調教用衣装:CHARA_ID:衣装部位_下着上 = CLOTH:CHARA_ID:衣装部位_下着上
調教用衣装:CHARA_ID:衣装部位_下着下 = CLOTH:CHARA_ID:衣装部位_下着下

;職業設定
;・無職の女王が生成されてかわいそうだったので設定。
TALENT:CHARA_ID:職業 = FINDELEMENT(JOB_NAME, "王族")

;初期経験を設定
;・話的に色々遊んでそうだし眷族一杯産んでそうなので多めに設定
種付数 = 20	;スキュラはココで設定した数値を基準にランダムで増える。設定値が8,15,50で取得技能が増える。
出産数 = 20	;現状出産側の処理が無効化されてるので意味はないが一応設定しておく。無効化されてる記述では設定値が3,8,20で取得技能が増える。
CALL RANDOM_CHARA_REPRODUCTION_SETTING(CHARA_ID, ,種付数, 出産数)
;↑の処理を参考に出産側をざっくり設定
TIMES 出産数, 1.6
出産数 += RAND:11
ABL:CHARA_ID:出産 = 10
EXP:CHARA_ID:出産経験 = 出産数
EXP:CHARA_ID:妊娠経験 = 出産数
SETBIT TALENT:CHARA_ID:淫乱系, 素質_淫乱_孕みたがり
SETBIT TALENT:CHARA_ID:淫乱系, 素質_淫乱_良質母胎

;慰安モブ設定を削除
;・↑RANDOM_CHARA_REPRODUCTION_SETTINGでモブ識別子を使用するので消すのは実行後
CFLAG:CHARA_ID:慰安モブ = 0
CSTR:CHARA_ID:モブ識別子 '= ""

;面識あり
CFLAG:CHARA_ID:面識 = 1

;所属設定
CFLAG:CHARA_ID:所属 = 所属_ゲストNPC

;口上設定
;CFLAG:CHARA_ID:汎用口上パターン = 20xx
;CFLAG:CHARA_ID:口上カラーコード = QUEST_42_SCYLLA_QUEEN_MESSAGE_COLOR

[IF_DEBUG]
PRINTFORMW DEBUG: スキュラの女王を[CHARA_ID:{CHARA_ID}]で作成しました
[ENDIF]
RETURN CHARA_ID

;-------------------------------------------------
; 「スキュラの女王」死亡時の処理
;-------------------------------------------------
;・家系に名前を残したいのでキャラデータは消さずに残しておく。
@QST_QUEST_42_SCYLLA_QUEEN_DEAD()
#DIM DYNAMIC CHARA_ID

;生成されていない場合は処理なし
CHARA_ID = NAME_TO_CHARA("QUEST_42_スキュラの女王")
SIF CHARA_ID == -1
    RETURN CHARA_ID

;特殊状態を死亡に設定
;・設定しておかないとターン終了時の妊娠判定等が発生する
CFLAG:CHARA_ID:特殊状態 = 特殊状態_死亡


code:
MAP_NAME = "海岸の探索"
MAP_IMAGE = "coast01.webp"
PATH_COLOR = "0x401008"
PATH_ALPHA = "0xff"
NODE_BASE_COLOR = "0x4007c"

path: ERB/QUEST/クエストデータ/QUEST_42_海岸の探索/海岸の探索.txt


code:
<DocumentElement>
  <GMAPDATA>
    <id>638465506330299747</id>
    <NODE_ID>0</NODE_ID>
    <NODE_NAME>海沿いの街</NODE_NAME>
    <POS_X>320</POS_X>
    <POS_Y>320</POS_Y>
    <PATH_LIST>1,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638465506650520506</id>
    <NODE_ID>1</NODE_ID>
    <NODE_NAME>ロングビーチ</NODE_NAME>
    <POS_X>300</POS_X>
    <POS_Y>220</POS_Y>
    <PATH_LIST>0,2,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638465506779047977</id>
    <NODE_ID>2</NODE_ID>
    <NODE_NAME>干潟の渡</NODE_NAME>
    <POS_X>340</POS_X>
    <POS_Y>200</POS_Y>
    <PATH_LIST>1,3,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638465506808927172</id>
    <NODE_ID>3</NODE_ID>
    <NODE_NAME>海神の上顎</NODE_NAME>
    <POS_X>320</POS_X>
    <POS_Y>160</POS_Y>
    <PATH_LIST>2,9,4,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638465506827910053</id>
    <NODE_ID>4</NODE_ID>
    <NODE_NAME>難破船</NODE_NAME>
    <POS_X>340</POS_X>
    <POS_Y>120</POS_Y>
    <PATH_LIST>3,5,7,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638465506925168239</id>
    <NODE_ID>5</NODE_ID>
    <NODE_NAME>岬への道</NODE_NAME>
    <POS_X>300</POS_X>
    <POS_Y>60</POS_Y>
    <PATH_LIST>4,6,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638465506954387945</id>
    <NODE_ID>6</NODE_ID>
    <NODE_NAME>灯台跡</NODE_NAME>
    <POS_X>180</POS_X>
    <POS_Y>80</POS_Y>
    <PATH_LIST>5,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638465507033928423</id>
    <NODE_ID>7</NODE_ID>
    <NODE_NAME>悪魔の岩礁</NODE_NAME>
    <POS_X>460</POS_X>
    <POS_Y>80</POS_Y>
    <PATH_LIST>4,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638465507082969356</id>
    <NODE_ID>8</NODE_ID>
    <NODE_NAME>イラス=ネレイス</NODE_NAME>
    <POS_X>520</POS_X>
    <POS_Y>80</POS_Y>
    <PATH_LIST />
  </GMAPDATA>
  <GMAPDATA>
    <id>638465507192248791</id>
    <NODE_ID>9</NODE_ID>
    <NODE_NAME>入り江</NODE_NAME>
    <POS_X>420</POS_X>
    <POS_Y>160</POS_Y>
    <PATH_LIST>3,10,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638465507220849302</id>
    <NODE_ID>10</NODE_ID>
    <NODE_NAME>入り江</NODE_NAME>
    <POS_X>500</POS_X>
    <POS_Y>200</POS_Y>
    <PATH_LIST>9,11,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638465507248188408</id>
    <NODE_ID>11</NODE_ID>
    <NODE_NAME>入り江の洞窟</NODE_NAME>
    <POS_X>520</POS_X>
    <POS_Y>260</POS_Y>
    <PATH_LIST>10,12,</PATH_LIST>
  </GMAPDATA>
  <GMAPDATA>
    <id>638465507268628282</id>
    <NODE_ID>12</NODE_ID>
    <NODE_NAME>海魔の巣窟</NODE_NAME>
    <POS_X>560</POS_X>
    <POS_Y>260</POS_Y>
    <PATH_LIST>11,</PATH_LIST>
  </GMAPDATA>
</DocumentElement>

path: ERB/QUEST/クエストデータ/QUEST_42_海岸の探索/海岸の探索.xml


code:
;-------------------------------------------------
;クエスト定義ファイル
;QUEST_43
;-------------------------------------------------
@QST_TITLE_QUEST_43
RESULTS '= "天才（災）魔導工学者フレーヴェ"

;ギルドメニューでクエストをフィルターする際のタグリスト。末尾の','を忘れないように
@QST_QUEST_TAG_43
RESULTS '= "単発,イベント,"

;-------------------------------------------------
;クエストの実行可能条件を定義
;-------------------------------------------------
@QST_AVAILABLE_QUEST_43(QUEST_ID)
#DIM QUEST_ID
SIF CFLAG:NAME_TO_CHARA("フレーヴェ"):所属 == 所属_プレイヤー
    RETURN 0
;前提としてエルのデイリーイベント「世界を学ぶ」を見ていることが必要
SIF K7_DFLAG_世界を学ぶ < 2
    RETURN 0
;前提として無限背塔の攻略が必要
SIF !QST_SV_QUEST_FLAG:15
    RETURN 0
RETURN 1

;-------------------------------------------------
;クエストのタイプ
;マップ探索を伴う場合 "MAP"
;文章のみで完結する場合 "DIALOGUE"
;-------------------------------------------------
@QST_TYPE_QUEST_43
RESULTS '= "DIALOGUE"

;クエストメニュー右上段の概要欄。10行まで
@QST_SUMALLY1_QUEST_43
RESULTS:0 '= "ある日、エルがあなたの元を訪ね、一通の手紙と紹介状を渡してきた"
RESULTS:1 '= "彼女が在学しているアカデミーの先輩の一人があなたに興味を持ち、会ってみたいという"
RESULTS:2 '= "エルが言うには「非常に優秀な方だけど、とても個性的でね……」らしいが……"

;クエストメニュー右下段の備考欄。難易度や条件、報酬等を記述。7行まで
@QST_SUMALLY2_QUEST_43
RESULTS:0 '= @"難易度：-"
RESULTS:1 '= ""
RESULTS:4 '= @"@C:{カラー_選択不可}@「ノルディアのアカデミーのフレーヴェ、ねぇ……」@/C@"
RESULTS:5 '= @"@C:{カラー_選択不可}@「あそこは奇人・変人の巣窟だが、とびっきりの変人らしいぞ」@/C@"

;-------------------------------------------------
;QST_INTRO_QUEST_n はQUEST開始前に呼ばれる
;クエストの導入文などを記述
;"DIALOGUE"型のクエストは当該関数のCALLのみ行う
;-------------------------------------------------
@QST_INTRO_QUEST_43
;SETCOLOR フレーヴェ_口上カラー
PRINTFORML %ANAME(MASTER)%はアカデミーの地下の一番奥の部屋をノックした
PRINTL そこは、研究室と呼ぶには随分と物々しく、いっそ拷問部屋や刑務所と呼んだ方がしっくりきそうだ
PRINTL 他の研究室と違い、ここだけ鋼鉄製の重厚なドアで閉ざされているからだ
PRINTW
PRINTL 紹介状を衛兵や受付に渡したとき、揃って珍獣でも見るような目つきで観察され、文面を読み終わった後にげんなりとした表情を浮かべられた
PRINTL ただ、割とすんなりと通されたところをみると、「先輩」は影響力の強い人物のようだ
PRINTW
PRINTL 「どうぞ。入りたまえ！」
PRINTL ドアの向こうから、少女のような声が聞こえてくる
PRINTL 地下の雰囲気には場違いな明るい声だ
PRINTW
PRINTFORML 「やあやあ。よく来てくれたね。君が%ANAME(MASTER)%だね？」
PRINTL 「本来ならもっと丁寧にもてなすべきなんだろうが、今手が離せなくてね。このままで失礼するよ」
PRINTFORML 目の前の少女──フレーヴェ・アイゼンは、%ANAME(MASTER)%を一瞥しただけで、すぐに手元の機械のようなものを弄り回し始めた
PRINTW
PRINTL フレーヴェ・アイゼン。人間とドワーフのハーフで、最年少でアカデミーに入学した
PRINTL 同じアカデミー出身の少女エルよりも幼く見えるが、彼女よりも一回り近く年上らしい
PRINTL エル曰く、「科学への情熱はアカデミー屈指」「アカデミーの破壊魔」「マイペースが服を着て歩いている」…
PRINTW
PRINTL 「エル君は息災かい？　手紙にはそう書かれていたが、実際に会っている君から見てどうだい？」
PRINTFORML %ANAME(MASTER)%はエルと出会ってからの出来事や隊商護衛での出来事などをフレーヴェに話した
PRINTL アカデミーを離れてショッキングな出来事も色々とあったが、一方で少しずつ成長し、世界を学んでいっていることも
PRINTW
PRINTL 「そうか、ならよかった。彼女は学園の中で埋もれていくには惜しいからね」
PRINTL 「非常に優秀な子だが、神経が細やかな所もあるからな。このアカデミーの中で生き残るなら多少の図太さやズルさも覚えた方がいい」
PRINTW
PRINTL ところで、フレーヴェはさっきからガラクタを弄り回して何をしているのだろうか？
PRINTL 「ガラクタなんて失礼な。これは世紀の大発明が実現しようとしているのだよ！」
PRINTL  フレーヴェは、複数のチューブにつながれた、なにやらごちゃごちゃした機械を手持ちのスパナでカンカンと叩く
PRINTW
PRINTL 「見たまえ！　この完璧なフォルム。これこそ、『超小型蒸気機関５号』だ！」
PRINTL ５号ということは、１～４号も製作されたのだろうか？
PRINTL 「彼らは尊い犠牲になってくれたよ」
PRINTW
PRINTL 「人類は蒸気機関を得て科学や生活水準が進歩した。だが、蒸気機関は巨大なものが多く、普及には今一つだ。機関車の運賃だって高いしな」
PRINTL 「だが！　ボクの考えた次世代型蒸気機関『超小型蒸気機関５号』は出力はそのままに低コスト、かつコンパクト！」
PRINTL 「これが普及すれば、馬車ではなくて一家に一台自家用車、なんて時代もやってくるのさ！」
PRINTW
PRINTFORML フレーヴェは熱弁をふるいながら、手元の超小型蒸気機関５号を叩く。音はカンカンからガンガンに変わり、%ANAME(MASTER)%は緊張する
PRINTL それ以上はいけないんじゃ──と言いかけたとき、超小型蒸気機関５号が火花を上げ始めた
PRINTFORML 「いかん！　%ANAME(MASTER)%、そこのバケツを持ってきてくれたまえ！」
PRINTW
PRINTFORML %ANAME(MASTER)%は言われるがまま、消火用のバケツの中身を超小型蒸気機関５号にぶちまける
PRINTL 出火は収まったものの、超小型蒸気機関５号はぶすぶすと白煙を上げている…
PRINTFORML 大惨事に至らなくてよかった。%ANAME(MASTER)%はほっと胸をなでおろす
PRINTW
PRINTL 「……まあ、科学に失敗はつきものだよ」
PRINTL 「理論は完璧なんだがなあ。多少の外部衝撃ではビクともしない素材がいるな……」
PRINTL 「そういえば、閉山した鉱山都市では珍しい鉱石が採れるという話だったが……試してみる余地はあるな……」
PRINTW
PRINTFORML そろそろここに来た目的を忘れてしまいそうなので、%ANAME(MASTER)%は思考に耽るフレーヴェに声をかける
PRINTL 「おっと、すまない。ボクの悪い癖でね。一度考えに集中すると止まらなくなるんだ」
PRINTW
PRINTL 「本題に入ろう。実はボクは、かねてから純粋なドワーフたちが存在していた南方大陸に興味があってね」
PRINTL 「都は大規模噴火によって消滅してしまったけれど……彼らは都以外にも、地下に様々な研究施設を建造していたようだ」
PRINTL 「レルム半島にある遺跡のように、古代の技術が眠っている可能性があるらしい。さまざまな文献を読み漁ったり、ドワーフたちに尋ねた結果、目星をつけてあるんだ」
PRINTW
PRINTL 「とはいえ、ボクは運動はからっきしでね。一人で行ったら辿り着く前に野盗に襲われて慰み者にされるか、野犬に襲われて餌になるかのどっちかだろう」
PRINTL 「そこで、可愛い後輩と親交のある君に護衛をお願いしたいんだが、どうかな？」
CALL ASK_YN("依頼を受ける","やめておく")
IF !RESULT
    PRINTL
    PRINTL 「そうか、そうか！　ボクの依頼を受けてくれたのは君が初めてだよ！」
    PRINTFORML フレーヴェは心底嬉しそうに%ANAME(MASTER)%の手を握り、ぶんぶんと上下に振る
    PRINTL 「君とは、長い付き合いになりそうだ。よろしく頼むよ……ああ、そういえば自己紹介を忘れていた、失礼」
    PRINTW
    PRINTFORML フレーヴェは%ANAME(MASTER)%に手を差し出し、自信に満ちた表情で自己紹介を始めた
    PRINTL 「ボクこそが、天才魔導工学者のフレーヴェ・アイゼンだ！　よろしく！」
    PRINTW
    CALL COLOR_PRINTW("―― 天才（？）魔導工学者のフレーヴェが仲間になった！", カラー_注意)
    PRINTL
    PRINTFORML 「おっと大事なことを言い忘れていた」
    PRINTFORML 「僕もエル君同様、君のところにしばらく厄介になるよ。後で研究に必要な機材とか持ってこさせるから、空き部屋をひとつよろしくね」
    PRINTFORML 「心配無用！　ちゃんと学長には許可を取ってあるから！　ああ、新しい体験だ、ワクワクするなあ！」
    PRINTW
    PRINTFORML 半ば強引に、フレーヴェは%CALLNAME:MASTER%の領地に住み着くことになった……
    PRINTFORML アカデミーは貴重な人材をホイホイ他所にやって大丈夫なのだろうか？
    PRINTFORML しかしエルといい、フレーヴェといい、自分の所には跳ねっかえりな娘と縁があるようだ……
    PRINTW
    CALL COLOR_PRINTW("―― 天才（？）魔導工学者のフレーヴェが住み込みになった！", カラー_注意)
    PRINTL
    CFLAG:NAME_TO_CHARA("フレーヴェ"):所属 = 所属_プレイヤー
    QST_SV_QUEST_FLAG:QST_有効クエストID = 1
ELSE
    PRINTL
    PRINTL 「そうか、残念だ。でも気が変わったらまた会いに来てくれたまえ」
    PRINTL 「次に来たときはお茶くらいは出すとしよう」
    PRINTL 「願わくば、エルが気に入っている君が依頼を受けてくれますように」
    PRINTW
ENDIF

CFLAG:NAME_TO_CHARA("フレーヴェ"):面識 = 1
TFLAG:クエスト_完了フラグ = 1
CALL TRAIN_EXIT()

path: ERB/QUEST/クエストデータ/QUEST_43_天才（災）魔導工学者フレーヴェ.ERB